Mathe mit Zellen

Neues Lernen mit Medien
im Mathematikunterricht der Sekundarstufe I

	A	B	C	D	E
1			Tarif mit Vertragsbindung		Prepaidtarif
2	monatlicher Grundpreis		24,95		0
3	Grundpreis über 2 Jahre		=C2*24		=E2*24
4	Anschaffung Telefon				
5	Anmeldegebühren				
6	Summe Fixkosten				
7	Fixkosten pro Monat				
8	monatliche Aufladung				
9	Sunshine				
10	Moonshine				
11	Sun local				
12	Moon local				
13	weekend				
14	Minutenpreis (Mittel)				
15	monatliche Telefonzeit				
16	0				
17	=A16+1				



Sehr geehrte Kolleginnen und Kollegen,

im Rahmen der Reihe „Wie weiter mit dem Mathematikunterricht?“ überreicht Ihnen das Amt für Schule die Handreichung „Mathe mit Zellen“ zum Mathematikunterricht in der Sekundarstufe I. Im Hinblick auf die Weiterentwicklung des Mathematikunterrichts mit dem Ziel eines mehr selbstständigen und aktiven Mathematiktreibens der Schülerinnen und Schüler werden Anregungen und Antworten gegeben auf die Frage eines sinnvollen Computereinsatzes unter Verwendung der Standardsoftware Tabellenkalkulation. Das „elektronische Rechenblatt“, die Tabellenkalkulation, kann in vielfältiger Weise und in einem breiten Spektrum mathematischer Themen sinnvoll eingesetzt werden. Nicht nur bei der Einführung neuer Unterrichtsinhalte sind heuristische Betrachtungen möglich. Steht im Mittelpunkt des Aufgreifens funktionaler Zusammenhänge im Mathematikunterricht die Frage „Was passiert, wenn...“, also die Grundfrage des für das Lernen von Mathematik fundamentalen operativen Prinzips, so arbeitet die gesamte Softwareklasse der Tabellenkalkulationen nach dem Prinzip des „was passiert, wenn...“. Durch den Einsatz der Software im Mathematikunterricht kann so das Entdecken, das Erarbeiten und das Herausstellen funktionaler Zusammenhänge erleichtert und verstärkt, ihre grafische Darstellung auf einfache Weise realisiert werden.

In der vorliegenden Handreichung hat der Verfasser versucht, ein im traditionellen Mathematikunterricht nicht unproblematisches Thema, nämlich den Umgang mit Variablen und funktionalen Zusammenhängen, durch den Einsatz einer Tabellenkalkulation in einer für Schülerinnen und Schüler interessanten und motivierenden Weise aufzubereiten. Seine Vorschläge zu einem sinnstiftenden und die Lerngruppe aktiv einbeziehenden Unterricht tragen dem Geist der neuen Rahmenpläne Mathematik Rechnung. Diese beschreiben einen Mathematikunterricht, der in einer Abfolge von Lernsituationen zum selbsttätigen Entdecken ermuntert und dabei den Einsatz des Computers im Allgemeinen und der Tabellenkalkulation im Besonderen ausdrücklich als Werkzeug und Inhalt zugleich vorsieht.

Zielgruppe der Handreichung sind die Lehrerinnen und Lehrer der Klassen 7 bis 10 des Gymnasiums, zum Teil auch der gymnasialen Oberstufe. Dennoch lassen sich die Prinzipien einer Arbeit mit Tabellenkalkulationen durch eine geeignete Auswahl von Aufgabenbeispielen bzw. die Wahl einfacher strukturierter Problemstellungen auf den Unterricht in anderen Schulformen übertragen. Auch muss nicht auf eine neue Generation von Schulbüchern gewartet werden. Enthält das eingeführte Lernbuch gute anwendungsorientierte Aufgaben, so ist gerade hier ein Einsatz angeraten. Ich ermuntere alle Kolleginnen und Kollegen, den hier aufgezeigten Weg selbst zu gehen und eigene Erfahrungen im Umgang mit dem Einsatz einer Tabellenkalkulation im Mathematikunterricht zu sammeln.

Dem Kollegen Peter Stender danke ich ausdrücklich für die geleistete Arbeit.

Werner Renz

Februar 2001

Herausgeber: Behörde für Bildung und Sport, Amt für Schule, Hamburg.

Satz: Amt für Schule, S 13/2

Druck: D & K Druck, Hamburg

Hamburger Schulen können die Handreichung von der Beschaffungsstelle V 243-2 beziehen.

Freie und Hansestadt Hamburg
Behörde für Bildung und Sport
Amt für Schule

Mathe mit Zellen

Neues Lernen mit Medien im Mathematikunterricht der Sekundarstufe I

Referat: Mathematisch-naturwissenschaftlich-technischer Unterricht
Werner Renz, S 13/2, Amt für Schule

Verfasser: Peter Stender, Gymnasium Oberalster

Inhaltsverzeichnis

Seite

1 Einleitung

'Lehrer sind Menschen, die uns helfen, Probleme zu lösen, die wir ohne sie nicht hätten'. Für kaum einen Unterrichtsinhalt gilt diese Einschätzung aus Sicht der Schülerinnen und Schüler wohl nachdrücklicher als für die Behandlung der Termumformungen im Mathematikunterricht der Sekundarstufe I. Ein Blick auf die Unterrichtssystematik zeigt, warum dies so ist. Termumformungen und ihr Nutzen werden erst Schuljahre später relevant, für die Schülerinnen und Schüler in den Klassen 7 und 8 bleibt der Sinn der Beschäftigung mit Termen weitgehend im Dunkeln. Die Schülerinnen und Schüler sind in diesen Klassenstufen in der Regel damit zufrieden, eine konkrete Rechnung mit Zahlen mit einem Ergebnis durchzuführen. Auch Beispiele, bei denen eine Fragestellung mittels Termumformung für eine größere Anzahl von Fällen auf einen Schlag beantwortet wird, können Schülerinnen und Schüler kaum von der Sinnhaftigkeit ihres Tuns überzeugen, da das Problem nicht ihr Problem ist und schon gar nicht die ganze Problemklasse.

Eine Antwort auf dieses fachdidaktische Dilemma könnte die Beschäftigung mit einer Tabellenkalkulation sein, und zwar ausgehend von verschiedenen Gesichtspunkten:

1. Tabellenkalkulationen stellen eine Standardcomputeranwendung in vielen Berufszweigen dar, und dies wird sich in den nächsten Jahren eher verstärken. Die unterrichtliche Behandlung dieser Software rückt also in den Bereich von 'Lebensvorbereitung im engeren Sinne' [Heymann]. Dies kann Schülerinnen und Schülern anhand von Beispielen und Aufgaben gut deutlich gemacht werden.
2. Die Verwendung von Computern im Unterricht ist für viele Schülerinnen und Schüler sehr motivierend.
3. Tabellenkalkulationen werden erst dann zu einem mächtigen Werkzeug, wenn man die Rechnungen mit Termen formuliert, da dann zusätzliche Eingaben, Änderungen oder wiederkehrende strukturgleiche Rechnungen automatisch umgesetzt werden.
4. Zu Beginn kann man Tabellenkalkulationen nutzen, ohne mit Termen umgehen zu können. Beim Umgang mit Tabellenkalkulationen geht man also natürlicherweise den Weg vom Rechnen mit Zahlen zum Umgang mit Termen.
5. Der Variablenbegriff wird auf einer sehr konkreten Grundlage eingeführt. Bei der Formulierung von Termen in der Tabellenkalkulation bezieht man sich immer auf konkrete Zellen, wobei dieser Bezug variabel umgesetzt wird, aber wiederum immer gleich numerisch ausgewertet wird. Damit wird von vornherein deutlich, dass eine Variable etwas ist, für das man viele verschiedene Zahlen einsetzen darf, auch wenn der Bezeichner immer gleich lautet.
6. Die Tabellenkalkulation findet auch schulintern fächerübergreifend gute Anwendungsmöglichkeiten, sei es bei der Auswertung von Messreihen im Physikunterricht, der Visualisierung von Daten in allen naturwissenschaftlichen Fächern oder bei der Bearbeitung statistischer Erhebungen, beispielsweise im Gemeinschaftskundeunterricht.

Beim Einsatz der Tabellenkalkulation tritt der Funktionsbegriff in den Vordergrund. Dabei werden nach und nach Terme zur Formulierung der Rechnungen herangezogen. Die Umsetzung einer Sachsituation in eine Rechenvorschrift ist ein wichtiger Schwerpunkt des Unterrichts.

Übungen zu den Termumformungen sind nicht Gegenstand dieser Handreichung, sie werden jedoch in Anschluss an den dargestellten Unterricht für Schülerinnen und Schüler deutlich sinnvoller.

2 Einführung in den Umgang mit einer Tabellenkalkulation

Für den Einsatz in der Mathematik in der Schule sind die gängigen Tabellenkalkulationsprogramme gleichermaßen geeignet. Verbreitet sind Works 3.0, StarOffice 4.0 oder Excel in den Versionen 5.0 bis Excel 2000. Die Bedienung erfolgt bei den vorgestellten Anwendungen in allen Programmen auf dieselbe Weise.

2.1 Handhabung

Die Eingaben in einer Tabellenkalkulation geschehen in Tabellen der folgenden Form:

B3		=4+5					
	A	**B**	**C**	**D**	**E**	**F**	**G**
1							
2	4		Text		42		
3	5	9					
4							
5					0,005	21	

Tabelle 1: Beispiele für Text, Zahlen und Rechnungen, **B3** *ist markiert*

2.1.1 Eingabe

Eine Eingabe in die Tabelle erfolgt, indem mit Hilfe von Cursortasten oder Maus die gewünschte Zelle markiert und dann der Inhalt über die Tastatur eingegeben wird. Dabei erscheint immer der Inhalt der markierten Zelle in der **Statuszeile** oberhalb der Tabelle. Ist die Eingabe fertig, beendet man sie mit `Return`, wodurch die Berechnung ausgelöst wird.

In die Tabelle sind verschiedene Arten von Eingaben erlaubt:

- Texteingaben zur Beschriftung von Zeilen, Spalten oder einzelnen Zellen der Tabellenkalkulation werden direkt in die Zelle geschrieben. Ist die rechts liegende Zelle leer, kann Text in diese hinein ragen, sonst muss man über das Menü einen Zeilenumbruch ermöglichen. Der Text kann wie in jeder Textverarbeitung gestaltet werden. (Zelle C2)
- Zahlen werden genauso eingegeben wie Text. Lässt sich eine Eingabe sinnvoll als Zahl interpretieren, so geschieht das automatisch. (Zellen A2, A3). Dezimalzahlen werden in deutschen Versionen mit Komma eingegeben. Ziffernfolgen mit Punkt werden – soweit sinnvoll möglich – als Datum interpretiert.
- Rechnungen erfolgen, wenn als erstes Zeichen in der Zelle ein '=' steht, gefolgt von einem berechenbaren Ausdruck (Zelle B3). Hierbei sind alle gängigen Rechenoperationen verfügbar. Die Inhalte anderer Zellen sind beim Rechnen verwendbar, indem man die Zellenbezeichnung eingibt. In B3 könnte also auch =A2+A3 stehen.
- Fallunterscheidungen sind gemäß dem folgenden Beispiel möglich:

  =WENN(E5<0,01;E2/2;E2) (Ergebnis in F5)

  Es sind drei Eingaben jeweils durch Semikolon getrennt. Die erste ist eine Bedingung, die zweite wird als Feldinhalt verwendet, wenn die Bedingung wahr ist, die dritte Eingabe, wenn die Bedingung falsch ist: =WENN(Bedingung;Wennfall;Sonstfall)

Ein bei Einsteigern oft auftretendes Problem bei der Eingabe in ein Feld der Tabellenkalkulation ist die Folge des Betätigens der Cursortasten links oder rechts. Dies führt nicht dazu, dass die Schreibmarke sich nach links bewegt, sondern wählt eine Zelle aus, deren Adresse an der aktuellen Stelle der Schreibmarke eingefügt wird. Dies führt bei Schülerinnen und Schülern oft zu Verwirrung. Betätigt man hingegen vorher die Taste `F2`, kann man die Schreibmarke wie in einer Textverarbeitung innerhalb der Zelle bewegen.

2.1.2 Verändern und Kopieren

Will man eine vorhandene Eingabe verändern, markiert man die betreffende Zelle und betätigt die Taste `F2`. Nun kann man den Zelleninhalt wie in einer Textverarbeitung bearbeiten.

Zum Kopieren des Inhaltes einer oder mehrerer Zellen in andere Zellen markiert man zunächst die zu kopierenden Zellen mit der Maus, analog der Blockmarkierung in einer Textverarbeitung. Dann kopiert man den markierten Inhalt in den Zwischenspeicher. Dies geschieht mit

- der Maus durch Anwahl des entsprechenden Punktes in der Menüleiste (Kopieren),
- Betätigen der Tastenkombination `CTRL-c` oder durch
- Betätigen der Tastenkombination `CTRL-Einfg`.

Nun markiert man mit den Maus den Zielbereich (es können mehrere Zellen sein) für den Kopiervorgang und fügt den Inhalt des Zwischenspeichers ein mit

- der Maus durch Anwahl des entsprechenden Punktes in der Menüleiste (Einfügen),
- Betätigen der Tastenkombination `CTRL-v` oder durch
- Betätigen der Tastenkombination `SHIFT-Einfg`.

Beim Kopieren von Rechnungen mit Zellbezeichnern werden diese verändert! Dies ist die wichtigste Stärke der Tabellenkalkulation:

	A	B	C	D	E	F	G
1	1						
2	=A1+1	=A2^2	=B2^2				
3	=A2+1	=A3^2	=B3^2				
4	=A3+1	=A4^2	=B4^2				
5	=A4+1	=A5^2	=B5^2				

Tabelle 2: Kopieren von Formeln

Die dargestellten Zellinhalte werden erzeugt, indem die Formeln in die Zellen A2 und B2 eingegeben werden und dann in die darunter liegenden Zellen kopiert werden. Die Spalte C wird erzeugt, indem die Spalte B nach rechts kopiert wird.

Eine didaktische Anmerkung hierzu ist im Abschnitt 2.2 nachzulesen.

Die Formeln in der Tabelle 2 werden sofort ausgewertet und führen zu folgender Ansicht:

	A	B	C	D	E	F	G
1	1						
2	2	4	16				
3	3	9	81				
4	4	16	256				
5	5	25	625				

Tabelle 3: Ausgewertete Tabelle 2

Der erläuterte Kopiermodus ist in allen Windows-Programmen identisch, also für Einsteiger mit entsprechenden Vorkenntnissen gut geeignet. Excel bietet die folgende Erleichterung:

	A	B	C	D	E	F	G
1	1						
2	=A1+1	=A2^2	=B2^2				
3	=A2+1	=A3^2	=B3^2				
4	=A3+1	=A4^2	=B4^2				
5	=A4+1	=A5^2	=B5^2				

Tabelle 4: Komfortabler Kopiermodus in Excel

Klickt man mit der Maus links auf das Quadrat rechts unten im markierten Bereich und zieht das Quadrat bei gedrückter Maustaste nach unten, werden die markierten Zellen nach unten kopiert. Dieses Verfahren lässt sich auch für das Kopieren in andere Richtungen anwenden, jedoch nur für eine Richtung zurzeit. Besonders komfortabel ist dies Verfahren, wenn man zwei untereinander stehende Zellen alternierend nach unten kopieren will:

	A	B	C	D	E	F	G
1	Start	0,003	Parameter	2			
2	Iteration						
3	=B1	0					
4	=A3	=D$1*A4*(1-A4)					
5	=B4	=B4					
6	=A5	=D$1*A6*(1-A6)					
7	=B6	=B6					
8	=A7	=D$1*A8*(1-A8)					
9	=B8	=B8					

Tabelle 5: Kopieren von alternierenden Zelleinträgen

Zieht man in der abgebildeten Situation das Quadrat mit gedrückter linker Maustaste nach unten, entstehen die gezeigten Zelleinträge. Dies ist bei zweistufigen Iterationen besonders nützlich.

2.1.3 Grafiken

Die erzeugten Zahlen können in Tabellenkalkulationen auf vielfältige Weise grafisch dargestellt werden. Dazu markiert man die darzustellenden Zellen inklusive etwaiger Zellen zur Beschriftung der Tabellen und wählt dann mit der Maus aus der Menüleiste den Eintrag für Grafiken aus. Nun muss in einigen Programmen mit der Maus der Bereich in der Tabelle festgelegt werden, in dem die Grafik erzeugt werden soll. Dann wird man durch ein Menü geführt, das die möglichen Darstellungsformen anbietet. Hierbei hat das Punktdiagramm Vorzüge gegenüber dem Liniendiagramm, da hier die Kontrolle über die x-Achse in Form der Daten in einer Spalte beziehungsweise Zeile beim Nutzer liegt.

Fertige Diagramme können bearbeitet werden. Bei Excel gelangt man über einen Doppelklick auf einen Diagrammteil (Graph, Hintergrundfläche, x- oder y-Achse, Rahmen, Beschriftung) in ein Menü zum Gestalten des jeweiligen Diagrammteils (Farbe, Muster, Achsenskalierung etc.). Ein Rechtsklick ins Diagramm erzeugt ein Menü mit dem Eintrag ‚Datenquelle', der es über ein Menü erlaubt, Datenreihen hinzuzufügen, zu löschen oder zu verändern.

Stellt man per Hand eingegebene Zahlen in einer Grafik dar, können die gezeichneten Punkte in der Grafik mit der Maus verschoben werden (Excel). Dies verändert den Eintrag in der Zelle. Mit diesem Vorgehen kann man auf einfache Weise Parameter steuern, die auf die Rechnung wirken: an erzeugt eine Grafik, die nur die Parameter enthält (nicht verbundene Punktgrafik mit deutlichen Punkten), und kann dann wie mit einem Rollbalken die ganze Rechnung steuern, die dann gegebenenfalls eine zweite Grafik steuert (Aufgabe 32).

2.2 Verschiedenartige Zellbezüge - eine didaktische Frage

2.2.1 Relative Spalten- und Zeilenbezüge

Die Adressierung von Zellen erfolgt in Tabellenkalkulationen auf zwei verschiedene Weisen:

– als relative Adresse, die beim Kopieren ihren Inhalt ändert, wie in den Beispielen oben, oder

– als absolute Adresse, die Kopiervorgänge unverändert übersteht. Hierzu muss der betreffende Teil der Adresse mit einem '$' versehen werden.

- $A$1 wird beim Kopieren nie geändert.
- Bei $A1 wird beim Kopieren nach unten die Zahl entsprechend der Kopierweite erhöht, das „A" bleibt jedoch auch beim Kopieren nach rechts erhalten.
- Bei A$1 wird beim Kopieren nach rechts der Buchstabe entsprechend der Kopierweite erhöht, die „1" bleibt jedoch auch beim Kopieren nach unten erhalten.

In älteren Tabellenkalkulationen wurde eine andere Schreibweise für die Adressierung verwendet, die in Excel noch einstellbar ist und einige didaktische Vorteile bietet:

Die Adressierung erfolgt sowohl in den Spalten als auch in den Zeilen über Zahlen:

- Z1S2 entspricht $A$2
- Z(-1)S bedeutet: gleiche Spalte aber eine Zeile höher als die Zelle, in der dieser Ausdruck steht.
- Z(3)S2 bedeutet: Spalte2 aber drei Zeilen tiefer.

- ZS(-1) bedeutet: gleiche Zeile aber eine Spalte links.
- Z1S(2) bedeutet: Zeile 1 aber zwei Spalten weiter rechts.

Diese Schreibweise hat erhebliche didaktische Vorteile:

- die Begriffe *absolute* und *relative Adressierung* werden bereits bei der Eingabe sinnvoll, nicht erst beim Kopieren,
- gleich bleibende Rechenverfahren in einer Spalte werden immer durch dieselbe Formel repräsentiert, was die Beschreibung mit einem Funktionsterm genau abbildet.

	1	2	3	4	5	6	7
1	1						
2	Z(-1)S+1	=ZS(–1)^2					
3	Z(-1)S+1	=ZS(–1)^2					
4	Z(-1)S+1	=ZS(–1)^2					
5	Z(-1)S+1	=ZS(–1)^2					

Tabelle 6: Alternative Zelladressierung

In dieser Darstellung ist die Zelladresse von der gleichen Struktur wie eine Variable in einem Funktionsterm. Daher ist diese Bezeichnung didaktisch vorzuziehen. Auch die Bedeutung von relativer und absoluter Adressierung sind hier viel einfacher zu verstehen, was für einen Einsatz dieser Schreibweise in der Schule spricht. Dagegen spricht jedoch die Tatsache, dass die zuerst beschriebene Adressierung sich zum Standard entwickelt und die alte nach meiner Kenntnis nur noch von Excel unterstützt wird.

Die Entscheidung, welche Adressierung im Unterricht gewählt wird, kann für jede Lerngruppe neu getroffen werden. Hier wird die Adressierung mit A2 etc. verwendet, um unabhängig von der Software des Lesers zu bleiben.

2.2.2 Bezeichnete Zellbereiche

Excel erlaubt es, einzelne Zellen, Spalten, Zeilen oder Zellblöcke mit einem Namen zu versehen, über den dann auf diese Zellen zugegriffen werden kann.

Anzahl							
	A	B	C	D	E	F	G
1			=x*Anzahl	1	=x^2		
2			=x*Anzahl	=D1+1	=x^2		
3		8	=x*Anzahl	=D2+1	=x^2		
4			=x*Anzahl	=D3+1	=x^2		
5			=x*Anzahl	=D4+1	=x^2		

Tabelle 7: Zellen mit Namen

Trägt man links oben statt der Beschriftung B3 einen anderen Text ein (z.B. Anzahl), so kann man auf die Zelle B3 künftig über Anzahl zugreifen, wobei dies stets als absolute Adresse interpretiert wird. In Tabelle 7 wurde ein Teil der Spalte D als x benannt und darauf in Spalte E zugegriffen. Dies wird jetzt als teils relative Adresse behandelt, so dass in E tatsächlich die Quadrate der Zahlen stehen, die in der gleichen Zeile in D stehen. Ebenso können Matrizen definiert werden, für die Excel einige Matrizenfunktionen (Aufgabe 52) zur Verfügung stellt.

Diese Bezeichnungsweisen sollten im Mathematikunterricht erst nach vollständigem Erwerb des Variablenbegriffes verwendet werden, da sonst die Gefahr besteht, dass ohne Verständnis für die Sache mit Buchstaben operiert wird. Der konkrete Zellbezug eines Ausdrucks wie D2 würde verschüttet.

3 Vom Zahlenrechnen zum Term

Haben viele Schülerinnen und Schüler einer Lerngruppe Erfahrung im Umgang mit dem Computer, beispielsweise einer Textverarbeitung unter Windows, sollte der Umgang mit der Tabellenkalkulation gleichzeitig mit der Behandlung des Funktionsbegriffs (siehe Kapitel 4) erarbeitet werden. Einige der hier genannten Aufgaben können dann zur Übung verwendet werden. Sind jedoch die grundlegenden Bedienschritte noch zu erlernen, wie der Umgang mit der Maus, den Cursortasten und das Kopieren von Ausdrücken, sollten diese mit den ersten Bedienschritten der Tabellenkalkulation zusammen erarbeitet werden. Günstig ist eine Einführung parallel mit dem Taschenrechner, da Probleme wie Klammersetzung, Vorzeichen und Struktur der Rechnung bei beiden Geräten gleichermaßen auftreten.

Beginnen sollte man mit der direkten Eingabe von Rechenaufgaben in verschiedene Zellen der Tabelle. Dabei kann jeweils ein Partner am Computer mit dem Taschenrechner nachrechnen, so dass mit beiden Medien zeitgleich geübt wird. Kleine Wettbewerbe ergeben sich hier fast von selbst. Die Beschriftung der Aufgaben dient zum Unterscheiden von Text und Formel. Aufgabenlisten, bei denen ähnliche Ausdrücke auftreten, sind ein Anlass, das Kopieren von Zellen und das Verändern von Zellinhalten zu erlernen, wobei zunächst ohne Zelladressierung gearbeitet wird.

Treten mehrfach Rechnungen auf, die sich nur in einer Zahl unterscheiden, ist es sinnvoll, diese Zahl in eine Zelle zu schreiben und auf diese Zahl über den Zellenbezug zuzugreifen. Dies muss vom Lehrervon der Lehrkraft vorgeführt werden. Als Beispiel mag die Umrechnung von Währungen dienen: ein Betrag in DM wird eingegeben und der zugehörige Dollarwert bestimmt. Diese Aufgabe kann derart ausgebaut werden, dass die Rechnung für mehrere Währungen nebeneinander erfolgt. Im nächsten Schritt sollten dann auch die Wechselkurse eingegeben werden können:

	A	B	C	D	E	F	G
1	Eingabe	125	DM				
2							
3		1	1	100	1	100	100
4		Euro	Dollar	Yen	Pfund	Schilling	Sfr
5	Kurs	1,95583	2,1585	1,9713	3,2494	14,2136	125,637
6	Ergebnis	=B3*B1/B5	=C3*B1/C5	=D3*B1/D5	=E3*B1/E5	=F3*B1/F5	=G3*B1/G5

Tabelle 8: Währungsumrechnung, Kurse vom 14.8.2000

Beginnen wird man mit den Spalten A und B, wobei B3 zunächst leer bleibt und die Formel in B6 ohne den entsprechenden Bezug steht. Werden dann weitere Spalten hinzugefügt, können die Einheiten in Zeile 3 zunächst direkt als Zahlen in die Formeln in Zeile 6 eingegeben werden. Wenn das Bedürfnis auftritt, in Zeile 6 überall einen einheitlichen Ausdruck zu erhalten, verwendet man den Zellbezug auf B3, C3 usw. Die Formulierung der Rechenanweisungen mit Zellbezügen entspricht der Beschreibung der Rechnung mit einem Term. Der Vorteil dieser Darstellungsweise ist in der Tabellenkalkulation offensichtlich. Kopieren die Schülerinnen und Schüler die Zellen mit den Formeln, so entdecken sie dabei wahrscheinlich die relative Adressierung, da sich die Zellbezüge dabei ändern. Dann kann an dieser Stelle die absolute Adressierung eingeführt und B1 zu B$1 verändert werden. Die folgenden Beispiele liefern weitere Anlässe.

Das Gauß-Problem („Addiere alle Zahlen von 1 bis ...“) ist auch mit der Tabellenkalkulation mühsam zu bearbeiten, wenn man die Aufgabe in einer Zelle eingeben will. Erzeugt man jedoch eine Zahlenreihe und addiert diese simultan, erhält man mit wenigen Eingaben und Kopierschritten eine Lösung.

	A	B	C	D	E	F	G
1	1	1		=A1*(A1+1)/2			
2	=A1+1	=A2+B1		=A2*(A2+1)/2			
3	=A2+1	=A3+B2		=A3*(A3+1)/2			
4	=A3+1	=A4+B3		=A4*(A4+1)/2			
5	=A4+1	=A5+B4		=A5*(A5+1)/2			
6	=A5+1	=A6+B5		=A6*(A6+1)/2			

Tabelle 9: Summenformel

Hier wird in Spalte D zum Vergleich der geschlossene Ausdruck berechnet. Die Zeilen 3 ff. werden wieder durch das Kopieren der darüberliegenden Zellen erzeugt, wobei die Zellbezüge automatisch angepasst werden. Zur weiteren Übung können danach die Summen über die ungeraden Zahlen, Quadrate oder Kuben bestimmt werden, wobei je nach Lerngruppe der geschlossene Ausdruck zum Beispiel über zu prüfende Hypothese behandelt werden kann.

Zum Üben des großen 1×1 sollen dieses aufgeschrieben werden. Die Schülerinnen und Schüler beginnen beispielsweise mit dem 1×12 und erstellen dann die Zeilen daneben:

	A	B	C	D	E	F	G
1		12	13	14			
2	1	=A2*12	=A2*13	=A2*14			
3	=A2+1	=A3*12	=A3*13	=A3*14			
4	=A3+1	=A4*12	=A4*13	=A4*14			
5	=A4+1	=A5*12	=A5*13	=A5*14			
6	=A5+1	=A6*12	=A6*13	=A6*14			

Tabelle 10: Großes 1×1

Zum Kopieren der Formeln nach rechts bedarf es der (teilweise) absoluten Adressierung: =A2*12 in B2 muss durch =$A2*12 ersetzt werden, damit das A beim Kopieren erhalten bleibt. Es muss dann noch die 12 durch die 13 ersetzt werden. Damit entsteht das Bedürfnis, nicht nur den ersten Faktor, sondern auch den zweiten Faktor über einen Zellbezug zu beschreiben. Dies führt dann zu:

	A	B	C	D	E	F	G
1		12	=B1+1	=C1+1	=D1+1	=E1+1	=F1+1
2	1	=$A2*B$1	=$A2*C$1	=$A2*D$1	=$A2*E$1	=$A2*F$1	=$A2*G$1
3	=A2+1	=$A3*B$1	=$A3*C$1	=$A3*D$1	=$A3*E$1	=$A3*F$1	=$A3*G$1
4	=A3+1	=$A4*B$1	=$A4*C$1	=$A4*D$1	=$A4*E$1	=$A4*F$1	=$A4*G$1
5	=A4+1	=$A5*B$1	=$A5*C$1	=$A5*D$1	=$A5*E$1	=$A5*F$1	=$A5*G$1

6	=A5+1	=$A6*B$1	=$A6*C$1	=$A6*D$1	=$A6*E$1	=$A6*F$1	=$A6*G$1

Tabelle 11: Großes 1×1 mit absoluten Adressen

Zunächst werden die erste Zeile und Spalte hergestellt. Dann wird die Multiplikation in B2 eingegeben und dieses Feld nach rechts und nach unten kopiert. Für die eigentliche Rechenvorschrift ist also nur die Eingabe <u>eines</u> Terms notwendig.

Ein kleines Anwendungsbeispiel ist die automatische Erstellung einer Rechnung inklusive Mehrwertsteuerberechnung. Dieses Problem kann beliebig komplex behandelt werden.

	A	B	C	D	E	F	G
1	Anzahl	Produkt	Einzelpreis incl. MWSt.	MWSt. enthalten	Betrag		
2	1	Brot	6,5	=B$7*C2/(1+B$7)	=A2*C2		
3	3	Milch	1,5	=B$7*C3/(1+B$7)	=A3*C3		
4	2	Butter	1,99	=B$7*C4/(1+B$7)	=A4*C4		
5	3	Nutella	2,7	=B$7*C5/(1+B$7)	=A5*C5		
6			=C2+C3+C4+C5		=E2+E3+E4+E5		
7	Steuersatz	0,07					

Tabelle 12: Rechnung

Dabei können je nach Lerngruppe noch weitere Bestandteile der Tabellenkalkulation behandelt werden:

- Statt =C2+C3+C4+C5 kann =SUMME(C2:C5) geschrieben werden.
- Es gibt ein Zellenformat ‘Währung’, das ‘DM’ an die Zahl anfügt, ohne dass die Zelle zu einer Textzelle wird. In einigen Tabellenkalkulationen kann statt ‘DM‘ auch ein eigener Text eingefügt werden, so dass alle Einheiten möglich sind. Hier werden Einheiten in der Regel nicht mit eingegeben , da sie in der Tabellenkalkulation als Zelleigenschaft auftreten.
- Es gibt ein Zellenformat ‘Prozent’, das 0,07 in der Form 7% darstellt. Eingegeben werden muss weiterhin 0,07 (nicht bei Office-97!).

4. Einführung des Funktionsbegriffs

4.1 Didaktische und methodische Überlegungen

4.1.1 Didaktik

Die Einführung von Funktionen in Klasse 7 geschieht in der Regel über Zuordnungen, wie sie ansonsten in der Mathematik nicht behandelt werden, nämlich indem Zuordnungsdiagramme (Pfeildiagramme) gezeichnet werden. Dies wird ergänzt durch Tabellen und Stabdiagramme oder Ähnliches. Dann werden als Einstieg in die Funktionen, für die ein Funktionsterm vorliegt, proportionale und teilweise umgekehrt proportionale Funktionen behandelt. Dieser Lernweg ergibt sich aus der Systematik der Mathematik: zunächst wird ein Begriff in großer Allgemeinheit eingeführt, dann mathematisch gefasst und anschließend werden seine Ausprägungen ausgearbeitet, wobei hierarchisch vom Einfachen zum Komplexen vorgegangen wird.

Für den Funktionsbegriff in der Schule bringt dieser Weg jedoch einige Probleme mit sich:

- Die zur Begriffsgründung behandelten Diagramme spielen in der folgenden Schulmathematik keine Rolle mehr. Die so gelegten Grundlagen zu einem für die Schülerinnen und Schüler greifbaren Funktionsbegriff geraten daher zwangsläufig schnell in Vergessenheit.
- Die erste behandelte Funktionsklasse, die proportionalen Funktionen, ist ein sehr spezieller Fall, mit dem es nicht gelingt, ein Bild von Funktionen im Allgemeinen zu repräsentieren. Aufgabenstellungen, die in diesen Bereich fallen, lassen sich besser als Dreisatz lösen. Nur in der Mittelstufenphysik spielen die proportionalen Funktionen eine wesentliche Rolle. Hier haben die Schülerinnen und Schüler jedoch keine Möglichkeit, das Besondere eines proportionalen Zusammenhanges zu erfassen, da ihnen kaum andere funktionale Zusammenhänge bekannt sind.

In Folge dieser beiden Probleme erwerben Schülerinnen und Schüler in der Regel kein Verständnis um den Funktionsbegriff. Funktionen werden sehr häufig mit Geraden gleichgesetzt. Und eine wesentliche Fähigkeit besteht offensichtlich darin, mit Hilfe zweier Zahlen, nämlich dem Steigungswert und dem y-Achsenabschnitt, eine Gerade zu zeichnen. Kommt dann später ein x^2 in der Funktionsvorschrift oder -gleichung vor, darf man sich nicht wundern, wenn Schülerinnen und Schüler auch weiter dazu neigen, wiederum mit Hilfe zweier Zahlen, dieses Mal den Parametern p und q, Parabeln zu zeichnen.

Die Einführung eines Begriffs in der Schule ist offensichtlich nicht mit der Methode der Mathematik (durch Bezug auf eine Definition) durchführbar, auch dann nicht, wenn vorbereitend einzelne Beispiele behandelt worden sind. Schon die Auffassung von dem, was ein Begriff ist, sollte nicht aus der Mathematik übernommen werden (ein Begriff ist, was in der Definition steht), sondern sich mehr an dem praktischen Umgang mit einem Begriff orientieren: einen Begriff zu beherrschen heißt semantisch, syntaktisch, fachsprachlich und inhaltlich korrekte Sätze mit ihm bilden zu können.

Zur Einführung des Funktionsbegriffs in der Schule bedarf es somit einer Vielzahl von verschiedenen Beispielen von Funktionen. Dabei müssen die Beispiele vom gleichen Typ sein, wie sie später im Mathematikunterricht auftreten, also Beispiele mit einer Rechenvorschrift und nicht

vom Typ 'Schüler → Sitzplatz' (eine Ausnahme bilden die Beispiele aus der Statistik). Damit die Schülerinnen und Schüler den Funktionsbegriff als sinnvoll erfahren, sollten diese Beispiele problemhaltigen Sachverhalten entnommen werden.
Im Sinne der mathematischen Definition ist eine Funktion ein Tripel aus Definitionsmenge, Wertemenge und Zuordnungsvorschrift. Es ist ebenfalls korrekt, eine Funktion als Menge geordneter Paare aufzufassen. Als Grundlage für den Unterricht hat die zweite Definition einige Vorteile:

- Definitionsmenge und Wertemenge spielen in den in der Schule behandelten Funktionstypen eine untergeordnete Rolle, praktisch relevant sind sie nur bei den gebrochenen Funktionen. Daher wird die Behandlung von Definitionsmenge und Wertemenge in der Schule – außer aus der Fachsystematik heraus – kaum begründbar. In der Definition über geordnete Paare treten Definitionsmenge und Wertemenge im Wesentlichen implizit auf, sie können also im Unterricht auch in den Hintergrund treten. Bei Anwendungsbeispielen kann ohne den begrifflichen Apparat der sinnvolle Definitionsbereich besprochen werden. Wenn dann gebrochen rationale Funktionen auftreten, kann man auf entsprechende praktische Vorerfahrungen zurückgreifen.
- Die Menge geordneter Paare wird repräsentiert durch die Wertetabelle (aufzählende Form der Beschreibung einer Menge) und den Graphen. Sie sind Darstellungen der Funktion. Dieser Zusammenhang ist mit der anderen Definition nicht so einfach und mathematisch korrekt auszudrücken.
- Formal korrekt taucht der Funktionsterm als zweite Koordinate auf: $(x, 3x+2)$. Dies unterstützt die Sprechweise vom Funktions**term**, im Gegensatz zur Funktions**gleichung**. Letzeres führt bei Schülerinnen und Schülern immer wieder dazu, dass sie mit einer Funktionsgleichung Umformungen durchführen, wie beim Lösen von Gleichungen.

Die Verwendung des Funktionsbegriffs als Menge geordneter Paare soll nur als mathematisch korrekter Hintergrund für den Unterricht verstanden werden, nicht als Unterrichtsgegenstand selbst. Als Sprechweise für den Unterricht bietet sich an, von der Beziehung zwischen zwei Größen zu reden. Dieser Funktionsbegriff zeigt, dass der Dreiklang 'Term – Tabelle – Graph' eine gute Leitidee für die Einführung des Funktionsbegriffes ist. Diese Leitidee wird durch die Verwendung der Tabellenkalkulation optimal unterstützt, da diese gerade auf der Grundlage von Tabellen mit Hilfe von Termen die Berechnungen durchführt und grafische Darstellungen auf einfache Weise ermöglicht. Dabei können durch den Einsatz einer Tabellenkalkulation Beispiele behandelt werden, die ohne Computernutzung am Rechenaufwand scheitern oder im Unterricht nur vereinzelt aufgenommen werden können.

Die Tabellenkalkulation bietet die Möglichkeit, Sachverhalte schrittweise zu mathematisieren, da ohne Aufwand Spalten mit Zwischenergebnissen erstellt werden, die jeweils einzelnen in der Problemstellung auftretenden Größen entsprechen. Nach und nach sollten jedoch im Sinne der mathematisch handwerklichen Lernziele mehrere Rechenschritte in einer Spalte zusammengefasst werden, um den Funktionsterm geschlossen zu erhalten und Termumformungsschritte zu erlernen. Da einzelne Übungen auch ohne Tabellenkalkulation durchgeführt werden, lernen die Schülerinnen und Schüler schnell den Vorteil einer kompakten Darstellung der Rechnung ken-

nen, insbesondere wenn moderne Taschenrechner vorhanden sind, in die ein Funktionsterm eingeben werden und für verschiedene Werte von x auswertet werden kann[1].
Wichtiger didaktischer Grundsatz der dargestellten Aufgaben ist, dass das Lernen von Mathematik von Problemstellungen ausgehen sollte. Die Schülerinnen und Schüler müssen im Unterricht erfahren, dass Mathematik mit dem Ziel betrieben wird, reale Fragestellungen zu beantworten, Sie sollte keine Wissenschaft sein, deren Sinn man nicht versteht, weshalb man den Anweisungen des Lehrers am besten ohne eigenes Denken folgt. Nur wenn Mathematik *angewendet* werden kann, gibt es eine Begründung für Mathematikunterricht. Auch allgemeine Bildungsziele können mit einem sich im Wesentlichen auf innermathematische Sachverhalte beziehenden Mathematikunterricht nicht erreicht werden, da dieser zu häufig zu rein mechanischem Abarbeiten von Aufgabentypen führt. Auch der Weg, zunächst die Mathematik und dann die Anwendungsaufgaben zu behandeln, fruchtet meist nicht. Die Behandlung der Mathematik durch die Schülerinnen und Schüler bleibt mechanisch; das angestrebte Lösen von Textaufgaben erfordert aber lebendiges auf der eigenen Vernunft basierendes Herangehen, so dass die Schülerinnen und Schüler sich durch Textaufgaben in der Regel überfordert sehen. Der Modellierungsprozess muss selbst erster Gegenstand von Mathematikunterricht sein. Dieser Modellierungsprozess ist keineswegs so gradlinig wie ihn Mathematiklehrer häufig vorführen (was ist ‚x', was wissen wir, welche Gleichung folgt daraus), sondern in der Praxis mit vielen Irrtümern und Sackgassen versehen. Man erinnere sich an die eigene Studentenzeit, wenn Aufgaben zu bearbeiten waren, deren Lösungen nicht in irgendeinem Buch standen. Für unsere Schülerinnen und Schüler haben einfache Textaufgaben den gleichen Komplexitätsgrad.

4.1.2 Methoden

Die hier vorgestellten Aufgaben gehen alle von einer inhaltlichen Fragestellung aus, erfordern also den oben beschriebenen Modellierungsprozess. Dies empfinden die Schülerinnen und Schüler erfahrungsgemäß als besonders schwierig. Daher ist es wichtig, dass sie zu der konkreten Aufgabe zunächst einige Fälle mit Zahlen durchprobieren. Zur Aufgabe 3 auf Seite 21 gehört dann beispielsweise diese Vorgehensweise:

Breite der Fläche	10 m	15 m	20 m
Rest für die Tiefe	40 m	35 m	30 m
Tiefe der Fläche	20 m	17,5 m	15 m
Umzäunte Fläche	200 m^2	262,5 m^2	300 m^2

Tabelle 13: Probieren mit Zahlen zu Aufgabe 3, Seite 21

Die Schülerinnen und Schüler sollten dabei die einzelnen Rechenschritte selbst entwickeln, ohne die Hürde zu haben, mit Variablen arbeiten zu müssen. Die Darstellung in der Tabelle ist in der Probierphase nicht möglich, da nicht von vorn herein gezielt die richtigen Rechenschritte durchgeführt werden. Am Ende der Probierphase sollte dann aber der Rechenweg systematisch, jedoch noch ohne Variablen, beschrieben werden können. Die so erarbeiteten Rechenschritte können dann in die Tabellenkalkulation und später in eine Termformulierung übernommen werden. Wird diese Herangehensweise bei der Bearbeitung von Textaufgaben zum System gemacht, so wird die Hemmschwelle beim Herangehen an Textaufgaben deutlich gesenkt. Dabei ist dieses Verfah-

[1] Beispielsweise Sharp EL-5646R oder vergleichbare Taschenrechner, die ansonsten nicht programmierbar und für den Mittelstufenunterricht gut geeignet sind.

ren nicht einmal unmathematisch. Es bleibt auch bei Fragestellungen, die selbst für einen ausgebildeten Mathematiker schwierig sind, die wichtigste Methode. Niemand kann einen wirklich komplexen Zusammenhang auf Anhieb als geschlossenen Term formulieren. Für unsere Schülerinnen und Schüler sind die üblichen Textaufgaben aus dem Mathematikunterricht regelmäßig komplexe Zusammenhänge.

Die vorgestellten Aufgaben können zum großen Teil in verschiedenen Klassenstufen mit unterschiedlichen mathematischen Mitteln bearbeitet werden:

- in den Klassen 7 und 8 mit einer Tabellenkalkulation durch systematisches Durchrechnen,
- in den Klassen 9 und 10 nach der Behandlung von „Quadratischen Funktionen" oder „Exponentialfunktionen" mit einfachen analytischen Mitteln und
- in der Oberstufe mit Hilfe der Differentialrechnung. Die Aufgaben ermöglichen es daher, inhaltliche Fragestellungen in den folgenden Jahrgängen wieder aufzugreifen und so für die Schülerinnen und Schüler inhaltliche Assoziationsmöglichkeiten zwischen den Jahrgängen zu schaffen.

Viele Aufgaben basieren auf geometrischen Sachverhalten, insbesondere auf der Berechnung von Flächen und Volumen. Es bietet sich daher an, diese Inhalte im Unterricht parallel zu behandeln, und gegebenenfalls dort, wo der Lehrplan einen Inhalt erst später vorsieht, Vorgriffe mit geringer mathematischer Tiefe zu machen, also ein Beispiel vorzuführen und die benötigte Rechenvorschrift in sprachlicher Form anzugeben.

Die angebotenen Aufgaben sind nicht als Lehrgang zu verstehen, der systematisch von vorn nach hinten durchgearbeitet wird, sondern als Sammlung, aus der man je nach Lerngruppe und Lernsituation sinnvoll auswählt. Bewusst stehen jedoch die einfachen Zusammenhänge am Ende, da sie als Spezialfall einer vielfältigeren Gruppe von Funktionen erkannt werden sollen. Bei der Auswahl sollte man beachten, dass einige Aufgaben Sequenzen bilden und teilweise einzelne Aufgaben zur Vorbereitung anderer dienen.

Die Aufgabensammlung dient der Einführung und Vertiefung des Funktionsbegriffs. Daher sollten die Aufgaben im Unterricht in der Weise nachbesprochen werden, dass den Schülerinnen und Schülern die Zusammenhänge Term–Tabelle–Graph deutlich werden. Zur Dokumentation ist es sinnvoll, ein Heft anzulegen, in dem die entstehenden Graphen skizziert werden. Die Schülerinnen und Schüler können dann den qualitativen Verlauf anhand der Zeichnung beschreiben. Dabei sollten sie auch die für die Aufgabenstellung wichtigen Teile der Beschriftung des Koordinatensystems (Achsenskalierung, Größen an den Achsen, Einheiten) eintragen. So entsteht im Laufe der Zeit eine Vorstellung von der Vielfalt der möglichen Graphen. Je nach Fortschritt der Lerngruppe kann man nach einer gewissen Zeit damit beginnen, den Funktionsterm geschlossen zu bestimmen. Für die bereits bearbeitet Aufgaben sollte dieser Schritt nachgeholt werden. So entsteht eine Sammlung von Graphen mit zugehörigen Funktionstermen. Zumindest Parabeln, Geraden und 'Sonstige' sollten unterschieden werden können, gegebenenfalls auch umgekehrt proportionale und kubische Funktionen. In leistungsstarken Lerngruppen können auch andere Funktionen auftauchen. Es ist sinnvoll, für diese Dokumentation ein Heft einzuführen, das nur für diesen Zweck verwendet wird. Dabei sollte jeweils eine Seite pro Funktion verwendet werden, mit Aufgabentext, Zeichnung, Tabellenkalkulationseinträgen und Funktionsterm. Wird ein hinreichend stabiles Heft verwendet, kann dieser Funktionenkatalog über mehrere Schuljahre als Nachschlagewerk verwendet werden.

Bei einigen Funktionen sollte man die Graphen über den von der Aufgabenstellung her sinnvollen Definitionsbereich hinaus von der Tabellenkalkulation darstellen lassen, um typische Verläufe als Ganzes sehen – beispielsweise bei Funktionen dritten Grades – und die sinnvolle Einschränkung des Definitionsbereiches thematisieren zu können.

4.1.3 Medien

Nur während eines Teils des Unterrichts mit Nutzung einer Tabellenkalkulation werden die Schülerinnen und Schüler selbst im Computerraum das Programm bedienen. Die ersten Erklärungen der Bedienung, die Nachbearbeitung von Aufgaben und die Vorbesprechung von Lösungsansätzen und Lösungsschritten wird auf einem zentralen Medium durchgeführt werden.

Optimal ist die Verwendung eines Computers mit Projektion des Bildschirminhaltes. Dieser ist in den meisten Computerräumen verfügbar. Die Aktionen des Lehrers können am Computer direkt verfolgt werden, und der Lehrer kann dabei mit Blick zur Klasse erklären. Für die Projektion gibt es zwei Techniken, ein auf den Overheadprojektor auflegbares Display oder ein Beamer, bei dem Projektionstechnik und Darstellung des Computerbildes in einem Gerät vereint sind. Beamer sind mobil einsetzbar und können mit in den Klassenraum genommen werden, wenn dort ein Computer oder Notebook zur Verfügung steht. Das Overheaddisplay ist in der Regel an einen Raum (meist den Computerraum) gebunden, da ein besonders lichtstarker Projektor benötigt wird, um akzeptable Projektionsergebnisse zu erreichen.

Verfügt man über Beamer und Notebook, lässt sich die Tabellenkalkulation auch im Unterrichtsverlauf für kurze Arbeitsabschnitte einsetzen. Es kann z.B. ein Schüler oder der Lehrer für ein behandeltes Nullstellenproblem das Newtonverfahren mit Hilfe der Tabellenkalkulation durchführen, so dass ohne langwierige und fehlerbehaftete Rechnungen eine Lösung erlangt wird.

In den Erarbeitungsphasen und Reflektionsphasen kann der Umgang mit der Tabellenkalkulation auch an der Tafel oder auf einer Overheadfolie (siehe Anlage auf S. 74) simuliert werden. Da hier überlegt werden muss, was beim Kopieren der Formeln geschieht, erfordert diese Arbeit eine bewusste Reflexion der eigenen Eingaben. Auf eine solche Sequenz sollte nicht verzichtet werden, da sie eher der Situation während einer Klassenarbeit ohne Computereinsatz entspricht.

4.2 Beispiele

4.2.1 Statistik

Die Behandlung der Statistik kann sowohl bei der Einführung in die Arbeit mit Tabellenkalkulationen als auch im Anschluss daran erfolgen.

Die einfachsten funktionalen Zusammenhänge für den Jahrgang 7 liefern statistische Betrachtungen, wie sie auch im Lehrplan für dieses Schuljahr vorgesehen sind. Daher ist es sinnvoll, diese Inhalte zu Beginn des Schuljahres zu behandeln und damit gleichzeitig eine Vorbereitung des Funktionsbegriffes zu erreichen.

Damit auch die zeichnerischen Fähigkeiten der Schülerinnen und Schüler weiter gefördert werden, sollten die Erhebung von Daten und deren Visualisierung nicht nur mit Hilfe einer Tabellenkalkulation durchgeführt werden. Sobald jedoch die Grundbegriffe wie Stabdiagramm, Häufigkeitspolygon etc. verfügbar sind, ist der Einsatz des Computers sinnvoll. Als Grundlage kann das Aufgabenmaterial des Schulbuches dienen, aber auch eigene statistische Erhebungen der

Schülerinnen und Schüler, z.B. Verkehrszählungen vor der Schule, Statistik über das Verkehrsmittel für den Schulweg aller Schüler einer Schule oder eine Erhebung der Schülermeinungen zur letzten Projektwoche.

Die Eingabe und Visualisierung der Daten selbst führt zu den grundlegenden Bedienelementen der Tabellenkalkulation. Die darauf basierende Bestimmung der Kenngrößen führt zu den ersten Termen. Die wesentlichen statistischen Funktionen gehören zwar zum Funktionsumfang jeder Tabellenkalkulation, sollten aber nicht verwendet werden. Sukzessives Aufsummieren und begleitendes Durchzählen zur Ermittlung der relativen Häufigkeiten liefern gegenüber der oben beschriebenen Einführung vergleichbare Lernschritte.

	A	B	C	D	E	F	G
1	Zähler	Daten	Summe	rel. Häuf.	Abstand vom Mittel	summiert	Mittel:
2	1	0,8	=B2	=B2/C$7	=ABS(B2-G$2)	=E2	=C7/A7
3	=A2+1	1,3	=B3+C2	=B3/C$7	=ABS(B3-G$2)	=E3+F2	
4	=A3+1	2,4	=B4+C3	=B4/C$7	=ABS(B4-G$2)	=E4+F3	mittlere Abweichung:
5	=A4+1	0,7	=B5+C4	=B5/C$7	=ABS(B5-G$2)	=E5+F4	=F7/A7
6	=A5+1	3,6	=B6+C5	=B6/C$7	=ABS(B6-G$2)	=E6+F5	
7	=A6+1	4,1	=B7+C6	=B7/C$7	=ABS(B7-G$2)	=E7+F6	

Tabelle 14: Statistik mit elementaren Rechenverfahren

An dieser Stelle erleben die Schülerinnen und Schüler nebenbei den sinnvollen und notwendigen Einsatz der Betragsfunktion, die sie sonst nie ernst nehmen. Normalerweise lassen sie Vorzeichen einfach weg, wenn sie es für erforderlich halten. Der Einsatz einer Funktion wird nicht benötigt.

Für die Betrachtung von Spannweite und Median muss nur die Datenreihe sortiert werden und dann auf die richtigen Zellen zugegriffen werden. Für Übungsaufgaben mit vorbereitetem statistischen Material können bei Verwendung der Tabellenkalkulation auch größere Datenmengen verarbeitet werden, wie sie in aktuellen amtlichen Statistiken beispielsweise des Statistischen Bundesamtes (http://www.statistik-bund.de) erhältlich sind.

4.2.2 Extremwertaufgaben

Fragestellungen, bei denen eine Funktion als Ganzes betrachtet werden muss, sind entweder Gesetzmäßigkeiten in der Physik oder Extremwertaufgaben. Für die Formulierung von Gesetzmäßigkeiten müssen Schüler bereits über ein ganzes Repertoire an Funktionen verfügen, während die Behandlung von Extremwertaufgaben mit Hilfe von Wertetabelle ohne weitere Vorkenntnisse möglich ist.

Aufgabe 1

Aus einem Stück DIN A4 Papier soll ein Karton ohne Deckel gebaut werden, der maximales Volumen hat[2]. Dies geschieht durch vierfaches Einschneiden des Papiers und geeignetes Kleben. Die Aufgabe kann durch Basteln vorbereitet werden.

	A	B	C	D	E	F	G
1	Blattlänge =	29,7	Blattbreite =	21			
2							
3	Höhe	Länge	Breite	Volumen			
4	0	=B$1-2*A4	=D$1-2*A4	=A4*B4*C4			
5	=A4+0,1	=B$1-2*A5	=D$1-2*A5	=A5*B5*C5			
6	=A5+0,1	=B$1-2*A6	=D$1-2*A6	=A6*B6*C6			
7	=A6+0,1	=B$1-2*A7	=D$1-2*A7	=A7*B7*C7			

Tabelle 15: Wertetabelle zur „Maximierung des Kartons“

In dieser Auflösung tritt das Maximum für eine Einschnittlänge von 4 cm auf. Verringert man die Schrittweite, so erhält man weitere Stellen hinter dem Komma. Vollzieht man diesen Schritt mehrfach, wird man die Schrittweite und die Startgröße jeweils in eine Zelle eintragen und sich in den entsprechenden Termen hierauf beziehen.



Abb. 1: Grafik zur Maximierung des Kartons

Die Herstellung der Grafik in dieser Form geschieht über ein Punktdiagramm mit verbundenen Punkten, wobei die Graphen zur Breite und Länge des Kartons gelöscht und die Farben nachbearbeitet wurden. Die Spalte der Höhe wurde zur Beschriftung der x-Achse verwendet.

Ein Vorteil in der Verwendung der Tabellenkalkulation liegt darin, dass die Modellbildung gut mit Zwischenergebnissen durchgeführt werden kann. Der Term für das Volumen wird zunächst nicht explizit entwickelt. Gerade zu Beginn der Unterrichtseinheit wird von den Schülerinnen

[2] Je nach Jahreszeit sowie Temperament der Lehrkraft und der Lerngruppe erfindet man eine geeignete Geschichte hierzu.

und Schülern dadurch ein geringeres Abstraktionsniveau erwartet, das dann langsam durch Integration mehrerer Teilterme in eine Zelle erhöht werden kann.

Aufgabe 2

Der Karton soll einen Deckel erhalten.



Abb. 2: Schnittmuster für den Karton mit Deckel

Dies ergibt die folgende Tabelle:

	A	B	C	D	E	F	G
1	Blattlänge	29,7	Blattbreite	21			
2	Anfang	0	Schrittweite	0,1			
3	Höhe	Länge	Breite	Volumen			
4	=B2	=(B$1-3*A4)/2	=D$1-2*A4	=A4*B4*C4			
5	=A4+D$2	=(B$1-3*A5)/2	=D$1-2*A5	=A5*B5*C5			
6	=A5+D$2	=(B$1-3*A6)/2	=D$1-2*A6	=A6*B6*C6			
7	=A6+D$2	=(B$1-3*A7)/2	=D$1-2*A7	=A7*B7*C7			

Tabelle 16: Karton mit Deckel

Nun liegt das Maximum bei 3,4 cm, es handelt sich wieder um ein Polynom dritten Grades.

Aufgabe 3

Mit einem Zaun der Länge 50 m soll an einem Fluss eine rechteckige Weidefläche abgeteilt werden. Wie sind Länge und Breite der Weidefläche zu bestimmen, damit diese möglichst groß wird.

	A	B	C	D	E	F	G
1	Zaunlänge =	50					
2	Anfang =	0	Schrittweite	0,1			
3	Feldbreite	Feldtiefe	Fläche				
4	=B2	=(B$1-A4)/2	=A4*B4				
5	=A4+D$2	=(B$1-A5)/2	=A5*B5				
6	=A5+D$2	=(B$1-A6)/2	=A6*B6				
7	=A6+D$2	=(B$1-A7)/2	=A7*B7				

Tabelle 17: Zaun am Fluss

In dieser Aufgabe ergibt sich im Gegensatz zu den Karton-Aufgaben eine Parabel. Die Unterschiede zu den Graphen beim Karton sind zwar gering, jedoch erkennbar und sollten im Klassengespräch herausgearbeitet werden.

Aufgabe 4

Eine Zündholzschachtel soll 5 cm lang sein und 45 cm^2 Inhalt haben. Bei welcher Breite und Höhe braucht man zur Herstellung am wenigsten Material? (aus [4])

	A	B	C	D	E	F
1		Schachtellänge in cm	5			
2		Volumen cm^3	45			
3	Höhe	Breite	Deckel	Front	Seite	Verbrauch in cm^2
4	0,1	=C$2/(C$1*A4)	=C$1*B4	=A4*B4	=C$1*A4	=3*C4+2*D4+4*E4
5	=A4+0,1	=C$2/(C$1*A5)	=C$1*B5	=A5*B5	=C$1*A5	=3*C5+2*D5+4*E5
6	=A5+0,1	=C$2/(C$1*A6)	=C$1*B6	=A6*B6	=C$1*A6	=3*C6+2*D6+4*E6
7	=A6+0,1	=C$2/(C$1*A7)	=C$1*B7	=A7*B7	=C$1*A7	=3*C7+2*D7+4*E7

Tabelle 18: Streichholzschachtel ohne Verschnitt

Es entsteht eine gebrochen rationale Funktion. Das Minimum liegt bei dieser Modellierung bei einer Höhe von 2,6 cm. Dabei sind in Spalte F verschiedene Modellierungen möglich, je nach Bauweise der Streichholzschachteln. Im Unterricht sollten verschiedene Beispiele für Streichholzschachteln behandelt werden, wobei auch die vorgegebenen Größen besser den vorliegenden Exemplaren entnommen werden. Gegebenenfalls sind Fronten oder Seiten der Schublade doppelt gelegt und Klebeflächen vorhanden. Die Front ist immer gleich groß. Dies sollte thematisiert werden, um die Schülerinnen und Schüler auf die Vorteile analytischer Gedankengänge aufmerksam zu machen.

Diese Modellierung kann kritisiert werden, indem man berücksichtigt, dass die zusammenhängenden Pappstücke aus einem Rechteck geschnitten werden:

	A	B	C	D	E	F	G
1		Schachtellänge in cm	5				
2		Volumen cm^3	45				
3	Höhe	Breite	Gesamtlänge innen	Gesamtbreite innen	Gesamtlänge außen	Gesamtbreite außen	Verbrauch in cm^2
4	0,1	=C$2/(C$1*A4)	=C$1+4*A4	=B4+4*A4	=C$1	=3*A4+2*B4	=C4*D4+E4*F4
5	=A4+0,1	=C$2/(C$1*A5)	=C$1+4*A5	=B5+4*A5	=C$1	=3*A5+2*B5	=C5*D5+E5*F5
6	=A5+0,1	=C$2/(C$1*A6)	=C$1+4*A6	=B6+4*A6	=C$1	=3*A6+2*B6	=C6*D6+E6*F6
7	=A6+0,1	=C$2/(C$1*A7)	=C$1+4*A7	=B7+4*A7	=C$1	=3*A7+2*B7	=C7*D7+E7*F7

Tabelle 19: Streichholzschachtel mit Verschnitt

Die optimale Streichholzschachtel ist nur noch halb so hoch, die entstehende Funktion nach wie vor gebrochen.

Aufgabe 5

Es soll ein vorn und hinten offenes Regal aus Brettern einer Sorte gebaut werden. Pro laufenden Meter kosten die 20 cm breiten Bretter 7 DM. Für die Holzteile stehen insgesamt 50 DM zur Verfügung. Wie sind die Außenmaße des Regals zu wählen, damit möglichst viel untergebracht werden kann (was durch das Volumen gemessen werden soll)? (nach [2])

Abb. 3: Regal

	A	B	C	D	E	F	G
1	Kosten =	50	Tiefe =	0,2			
2	Preis/m =	7	Gesamtlänge =	=B1/B2			
3	Breite	Regale	Seiten	Volumen			
4	0	=6*A4	=(D$2-B4)/2	=A4*C4*D$1			
5	=A4+0,1	=6*A5	=(D$2-B5)/2	=A5*C5*D$1			
6	=A5+0,1	=6*A6	=(D$2-B6)/2	=A6*C6*D$1			
7	=A6+0,1	=6*A7	=(D$2-B7)/2	=A7*C7*D$1			

Tabelle 20: einfaches Regal

Es entsteht eine Parabel mit dem Maximum bei einer Breite von etwa 60cm. Das genaue Ergebnis lautet $\frac{25}{42}$ cm = 0,5952... cm. Dieses Ergebnis kann zum Anlass genommen werden, Näherungslösungen und eine sinnvolle Genauigkeit zu thematisieren. Will man dies an dieser Stelle nicht, kann man beispielsweise die Aufgabe mit 50,40 DM beginnen und erhält exakt 60cm für die Regalbreite. Auch in den folgenden Aufgaben sind die angegebenen Regalbreiten gerundet.

Aufgabe 6

Die Aufgabe kann modifiziert werden unter Verwendung anderer Regaltypen:



Abb. 4: Doppelregal



Abb. 5: Regal mit stärkeren Wänden

Für das Doppelregal sind nur geringfügige Veränderungen an der Tabelle durchzuführen:

	A	B	C	D	E	F	G
1	Kosten =	50	Tiefe =	0,2			
2	Preis/m =	7	Gesamtlänge =	=B1/B2			
3	Breite	Regale	Seiten	Volumen			
4	0	=6*A4	=(D$2-B4)/3	=A4*C4*D$1			
5	=A4+0,1	=6*A5	=(D$2-B5)/3	=A5*C5*D$1			
6	=A5+0,1	=6*A6	=(D$2-B6)/3	=A6*C6*D$1			
7	=A6+0,1	=6*A7	=(D$2-B7)/3	=A7*C7*D$1			

Tabelle 21: Doppelregal

Interessanterweise liegt auch hier das Maximum bei einer Gesamtbreite des Regals von 60 cm. Man sollte also bei diesem Regal für sinnvolle Regalbreiten mehr Geld investieren. Durch Probieren erkennen die Schülerinnen und Schüler schnell, dass diese Breite, bei der das maximale Volumen auftritt, unabhängig von der Anzahl der Seitenteile ist. Dies kann man beim Vorliegen der mathematischen Voraussetzungen analytisch klären.

Der Funktionsterm lautet $\text{Volumen} = \text{Tiefe} \cdot \frac{\text{Breite} \cdot (\text{Kosten} - \text{Meterpreis} \cdot \text{Breite})}{\text{Anzahl der Seitenteile}}$

Durch Probieren mit verschiedenen Anzahlen von Seitenteilen kann in Jahrgang 7 zumindest die Einsicht erreicht werden, dass dieser Faktor im Nenner nur das Volumen aber nicht die Größe der optimalen Breite ändert. Mit leistungsstärkeren Lerngruppen ist die Erklärung mit Hilfe des Funktionsterms durchführbar. Die Schülerinnen und Schüler erkennen, dass durch Termbearbeitung weitere Einsichten möglich sind. Ebenso kann zumindest empirisch geklärt werden, wie die Gesamtlänge, also letztlich die Kosten, und die Anzahl der Regale die Größe der optimalen Breite beeinflusst. Da Probieren mit Hilfe der Tabellenkalkulation schnell geht, können gut einfache Hypothesen gebildet und überprüft werden.

Aufgabe 7

Der Regaltyp mit den verstärkten Seitenteilen erfordert eine neue Analyse, da das dickere Holz teurer ist (hier 8 DM).

	A	B	C	D	E	F	G
1	Kosten =	50	Tiefe =	0,2			
2	Preis/m=	7	Preis/m =	8			
3	Breite	Regale	Regalkosten	Seitenkosten	Seite	Volumen	
4	0	=6*A4	=B4*B$2	=B$1-C4	=D4/D$2/2	=A4*E4*D$1	
5	=A4+0,1	=6*A5	=B5*B$2	=B$1-C5	=D5/D$2/2	=A5*E5*D$1	
6	=A5+0,1	=6*A6	=B6*B$2	=B$1-C6	=D6/D$2/2	=A6*E6*D$1	
7	=A6+0,1	=6*A7	=B7*B$2	=B$1-C7	=D7/D$2/2	=A7*E7*D$1	

Tabelle 22: Regal mit verstärktem Seitenteil

Wieder ergibt sich bei gleichen Kosten und gleicher Regalzahl die optimale Breite von etwa 60 cm. Der Funktionsterm ist nur geringfügig aufwendiger.

Aufgabe 8

Die Regalaufgabe kann auch mit vorgegebenem Volumen statt vorgegebenen Kosten gerechnet werden. Bei $0{,}2\text{m}^2$ ergeben sich dann wieder die bekannten 60 cm als Optimum.

	A	B	C	D	E	F	G
1	Volumen =	0,2	Tiefe=	0,2			
2	Preis/m=	7					
3	Breite	Höhe	Kosten				
4	0,1	=B$1/(D$1*A4)	=B$2*(6*A4+2*B4)				
5	=A4+0,1	=B$1/(D$1*A5)	=B$2*(6*A5+2*B5)				
6	=A5+0,1	=B$1/(D$1*A6)	=B$2*(6*A6+2*B6)				
7	=A6+0,1	=B$1/(D$1*A7)	=B$2*(6*A7+2*B7)				

Tabelle 23: Regalkosten bei vorgegebenem Volumen

Es entsteht jetzt keine Parabel mehr, sondern der Graph einer gebrochen rationalen Funktion.



Abb. 6: Kosten bei vorgegebenem Volumen

Aufgabe 9

Ein Bauunternehmer will drei Reihenhäuser mit flachem Dach und einer Grundfläche von jeweils $72\ \text{m}^2$. Dabei sollen die Außenwände und die Wände zwischen den Häusern mit möglichst geringem Aufwand erstellt werden. Diese Wände sind alle gleich teuer. Wie sollen Länge und Breite der Reihenhäuser gewählt werden?



Abb. 7: relevante Mauern der Reihenhäuser

Diese Aufgabe entspricht vollständig Aufgabe 8. Es entsteht eine gebrochene Funktion mit der optimalen Grundfläche $10\ \text{m} \times 7{,}2\ \text{m}$.

Gibt man beispielsweise die Kosten für die Mauern und damit deren Gesamtlänge vor, erhält man wieder eine Parabel.

	A	B	C	D	E	F	G
1	Fläche	72					
2			Gesamtlänge				
3	Breite	Länge	Mauern				
4	1	=B$1/A4	=4*A4+6*B4				
5	=A4+1	=B$1/A5	=4*A5+6*B5				
6	=A5+1	=B$1/A6	=4*A6+6*B6				
7	=A6+1	=B$1/A7	=4*A7+6*B7				

Tabelle 24: Reihenhaus

Aufgabe 10

Beim Bau eines Geschäftshochhauses werden die Kosten für den Bauplatz und die Planung mit 1,4 Millionen DM veranschlagt, das Erdgeschoss mit 200 000 DM. Jede weitere Etage wird 10 000 DM teurer als die vorherige. Später wird jede Etage monatlich 10 000 DM an Miete einbringen. Wie viele Etagen sind zu bauen, wenn ein möglichst günstiges Verhältnis von Baukosten zu Mieteinnahmen erzielt werden soll? (aus [4]).

	A	B	C	D	E
1	Planung	1400000			
2		Kosten pro Geschoss	Gesamtkosten	Einnahmen	Verhältnis
3	Erdgeschoss	200000	=B3+B1	10000	=D3/C3
4	1	=B3+10000	=C3+B4	=D3+10000	=D4/C4
5	=A4+1	=B4+10000	=C4+B5	=D4+10000	=D5/C5
6	=A5+1	=B5+10000	=C5+B6	=D5+10000	=D6/C6
7	=A6+1	=B6+10000	=C6+B7	=D6+10000	=D7/C7

Tabelle 25: Bau eines Geschäftshochhauses

Das Optimum liegt bei 16 Geschossen, man benötigt fünf Nachkommastellen im Verhältnis, um die Frage zu beantworten. Die Funktion in Spalte E ist gebrochen rational.

Aufgabe 11

Auf einem Baugrundstück, das die Form eines rechtwinkligen Dreiecks mit den Kathetenlängen 80 m und 100 m hat, soll eine Halle mit rechteckigem Grundriss und maximaler Grundfläche entstehen. Wie sind die Abmessungen der Halle zu wählen? (aus [1])



Abb. 8: Baugrundstück mit Gebäude

	A	B	C	D	E	F	G
1	1. Kante	2.Kante	Schrittweite				
2	100	80	0,8				
3	Hallenlänge	Hallenbreite	Fläche				
4	0	80	=A4*B4				
5	=A4+1	=B4-C$2	=A5*B5				
6	=A5+1	=B5-C$2	=A6*B6				
7	=A6+1	=B6-C$2	=A7*B7				

Tabelle 26: Dreieckiges Baugrundstück für Halle

Wieder ergibt sich eine Parabel mit dem maximalen Flächeninhalt in der Konfiguration (A2/2;B2/2). Diese Aufgabe kann mit anderen geradlinig begrenzten Grundstücken variiert werden.

Bringt eine andere Anordnung des Gebäudes einen Vorteil oder einen Nachteil? Zur Beantwortung dieser Frage kommt man mit einfachen geometrischen Überlegungen aus: „Faltet“ man die nicht bebaute Fläche über die bebaute, stellt man fest, dass im optimalen Fall beide gleich groß sind, unabhängig von der Kante, an der das Gebäude liegt, solange im Grundstück kein Winkel über 90° existiert.



Abb. 9: Baugrundstück mit Gebäude

4.2.3 Aufgaben mit linearer Interpolation

Aufgabe 12

Ein Fußballstadion hat insgesamt 33 Sitzreihen. In der untersten Reihe befinden sich 800 Sitzplätze, in der obersten 4160. Die Anzahl der Plätze nimmt von Reihe zu Reihe um den gleichen Betrag zu. Wie viele Sitzplätze hat das Stadion? (aus [1]).

	A	B	C	D	E	F	G
1	1. Reihe:	800	33. Reihe soll:	4160			
2	Zuwachs:	105	33. Reihe ist:	=B37			
3		Gesamtzahl Sitzplätze:	=C37				
4	Reihe	Plätze in Reihe	Summe Plätze				
5	1	800	=B5				
6	=A5+1	=B5+B$2	=C5+B6				
7	=A6+1	=B6+B$2	=C6+B7				

Tabelle 27

Die wichtige zunächst unbekannte Größe in dieser Aufgabe ist B2. Durch Probieren dieser Schrittweite erreicht man, dass in D1 und D2 die gleichen Werte stehen und kann dann die Gesamtzahl der Plätze ermitteln. Dabei ist natürlich anzustreben, dass die Schülerinnen und Schüler versuchen, diese Zahl rechnerisch zu ermitteln, um die Idee der Steigung einer Geraden vorzubereiten. Die Graphen der Funktion sollten gezeichnet werden, auch wenn sie für die Beantwortung der Aufgabenstellung nicht erforderlich sind, um die Qualität der Zusammenhänge zu erkennen. Die folgenden Aufgaben stellen Variationen dieser Fragestellung dar.

Aufgabe 13

Die Wand des abgebildeten Blockhauses (Breite 4,5 m) soll aus 20 cm dicken Stämmen hergestellt werden. Wie viele laufende Meter Holz werden insgesamt verbraucht? Wie groß ist die Fläche der Wand ungefähr?



Abb. 10: Blockhaus

	A	B	C	D	E	F	G
1	1. Balken	0	9. Balken soll	4,5			
2	Zuwachs	0,5	9.Balken ist	=B14			
3	Fläche:	=D3*0,2	Gesamtlänge:	=C14+13*4,5			
4	Balken	Balkenlänge	Gesamtlänge Dachbalken				
5	0	=B1	=B5				
6	=A5+1	=B5+B$2	=C5+B6				
7	=A6+1	=B6+B$2	=C6+B7				

Tabelle 28: Blockhauswand mit Dachschrägen

Bei der ermittelten Fläche handelt es sich nur um eine Näherung (in Form einer Riemannschen Untersumme), was an dieser Stelle thematisiert werden kann, aber für den Unterrichtsgang selbst nicht erforderlich ist.

Aufgabe 14

Eine 6,24 m lange Wand in einem „Nurdachhaus“ soll mit einer musterlosen Tapete tapeziert werden. Die Wand hat die Form eines gleichschenkligen Dreieckes und ist in der Mitte 4 m hoch. Die Tapetenbahnen sind 53 cm breit. Wie viele Rollen Tapete muss man kaufen, wenn eine Rolle 33,5 m lang ist (Standardmaß Rauhfasertapete)? Wie groß ist die Fläche ungefähr? Dabei sollte es geduldet werden, dass eine Bahn aus zwei Stücken Tapete tapeziert wird.

	A	B	C	D	E	F
1	1. Bahn	0	6. Bahn soll	4		
2	Zuwachs	0,6666666	6. Bahn ist	=B11		
3	Gesamt:	=C11*2	Rollen:	=AUFRUNDEN(B3/33,5;0)		
4	Bahn	Bahnlänge	Ges.länge Bahnen	Fläche:	=B3*0,52	
5	0	=B1	=B5			
6	=A5+1	=B5+B$2	=C5+B6			
7	=A6+1	=B6+B$2	=C6+B7			

Tabelle 29: Tapezieren

Bei dieser Aufgabe wird der Vorteil der Berechnung von B2 deutlich, da Probieren erst nach vielen Versuchen zum Ziel führt. Dieser Effekt kann auch bei den anderen Aufgaben erzeugt werden. Der Befehl AUFRUNDEN(B3/10,6;0) muss nicht unbedingt verwendet werden, wenn man stattdessen einige Nachkommastellen mit anzeigt und den eigenen Kopf benutzt.

Die Aufgabe kann ergänzt werden, indem man die Problematik der Mustertapete mit Verschnitt aufgreift.

Aufgabe 15

Ein Feld der abgebildeten Form wird von einem Mähdrescher mit der Mähbreite 12 m in senkrechten Bahnen gemäht.

Welche Strecke legt der Mähdrescher dabei zurück und wie groß ist näherungsweise die gemähte Fläche?



Abb. 11: Feld für Mähdrescher

	A	B	C	D	E	F	G
1	1. Bahn	780	161. Bahn soll	2780			
2	Zuwachs	=2000/160	161. Bahn ist	=B165			
3	Bahnlänge Trapez plus Rechteck:	=C164+2780*80	Fläche:	=B3*12			
4	Bahn	Bahnlänge	Gesamtlänge Bahnen				
5	1	=B1	=B5				
6	=A5+1	=B5+B$2	=C5+B6				
7	=A6+1	=B6+B$2	=C6+B7				

Tabelle 30: Mähdrescher

4.2.4 Extremwertaufgaben mit Interpolation

Aufgabe 16

Ein Verlag will ein wissenschaftliches Buch herausbringen und dafür den optimalen Preis festsetzen. Es stehen die folgenden Informationen zur Verfügung:

Für das Setzen des Buches, für Zeichnungen, Verhandlungen mit dem Autor usw. entstehen Grundkosten in Höhe von 12000 DM. Pro Exemplar entstehen zusätzlich Herstellungskosten (Druck, Papier) von 10 DM.

Aufgrund einer Abnehmerbefragung und von Erfahrungen mit früher erschienenen und ähnlichen Büchern kann davon ausgegangen werden, dass das Buch von 10000 Käufern erworben würde, wenn es der Verlag mit 10 DM abgibt (der Ladenpreis ist entsprechend höher), und von 4000 Käufern bei einem Abgabepreis von 20 DM (nach [2]).

	A	B	C	D	E	F	G
1		Vorkosten	12000				
2		Weite	1000				
3		Stückkosten	10				
4		Weite Preis	=10/6000				
5	Stück	Druckkosten	Gesamtkosten	Preis	Einnahmen	Gewinn	in %
6	10000	=A6*C$3	=B6+C$1	10	=A6*D6	=E6-C6	=F6/C6
7	=A6-C$2	=A7*C$3	=B7+C$1	=D6+C$2*C$4	=A7*D7	=E7-C7	=F7/C7

Tabelle 31: Verlag

Diese Aufgabe enthält mehrere Schwierigkeiten. Der Ansatz für den Verkaufspreis des Buches muss wie in den vorhergehenden Aufgaben durch Interpolieren zwischen 4000 und 10000 Exemplaren ermittelt werden. Für lineares Interpolieren empfiehlt es sich, wie in Tabelle 31 vorgeschlagen, von 10000 aus abwärts zu rechnen, da man hierbei bis auf Null Exemplare herunter kommt und die höchste Rendite von 55,3% bei 3300 verkauften Exemplaren liegt. Die Aufgabe wäre einfacher zu implementieren, wenn der Term für den Preis in Abhängigkeit von der Stückzahl ($26\frac{2}{3} - \frac{x}{600}$) analytisch vorliegen würde. Ganz ohne analytische Betrachtungen ist der Wert in C4 wieder durch Probieren zu ermitteln. Die weiteren Schwierigkeiten betreffen weniger mathematische Fragen, als die Frage, welche Zeile für den Verlag optimal ist: maximaler Gewinn (Spalte F), maximale Rendite (Spalte G) oder maximaler Umsatz (Spalte E)? Ist die lineare Interpolation sinnvoll oder gibt es bessere Ansätze?

Für den maximalen Gewinn ergibt sich bei linearer Interpolation eine Parabel.

Aufgabe 17

Eine Elektronikfirma verkauft monatlich 2000 Taschenrechner zum Stückpreis von 40 DM (Verkaufspreis an die Händler). Eine Marktforschung hat ergeben, dass der Umsatz immer dann um 300 Stück steigt, wenn der Preis um eine Mark sinkt. Herstellung und Vertrieb kosten 20 DM pro Taschenrechner. Welches ist der optimale Preis für die Taschenrechner?

	A	B	C	D	E	F	G
1		Stückkosten	20				
2		Schrittweite	300				
3	Stück	Preis	Einnahmen	Ausgaben	Gewinn	Rendite	
4	2000	40	=A4*B4	=A4*C$1	=C4-D4	=E4/D4	
5	=A4+C$2	=B4-1	=A5*B5	=A5*C$1	=C5-D5	=E5/D5	
6	=A5+C$2	=B5-1	=A6*B6	=A6*C$1	=C6-D6	=E6/D6	
7	=A6+C$2	=B6-1	=A7*B7	=A7*C$1	=C7-D7	=E7/D7	

Tabelle 32: Taschenrechner

Diese Aufgabe ist einfacher als die vorherige. Hier ist die Rendite eine lineare Funktion des Umsatzes mit negativer Steigung. Der Gewinn ist wieder eine quadratische Funktion.

Aufgabe 18

Auf einem Schulfest will eine Klasse Grillwürstchen verkaufen. Die Würstchen kosten im Einkauf 80 Pf pro Stück. Die Klasse überlegt, wie teuer sie die Würstchen verkaufen soll, um möglichst viel Gewinn zu machen. Die Einnahmen sollen für die Renovierung und den Ausbau der Schulcafeteria verwendet werden.

In dieser Aufgabe[3] steht der Modellbildungsprozess der Nachfragefunktion im Vordergrund und kann zu einem kleinen Mathematikprojekt ausgedehnt werden. Die Rechenvorschrift für den Gewinn, verkaufte Stückzahl∃(Verkaufspreis – Einkaufspreis)‘ ist schnell gefunden, die Frage, wie die verkaufte Stückzahl vom Verkaufspreis abhängt, kann vielfältig modelliert werden. Zunächst wird nur ein einfacher Ansatz gewählt, der dann verbessert wird. Die auftretenden Abschätzungen selbst müssen für jede Schule gesondert durchgeführt werden.

- Werden die Würstchen verschenkt, so ergibt sich eine Obergrenze für den möglichen Würstchenumsatz. Man könnte erwarten, dass durchschnittlich jeder Besucher des Festes zwei Würstchen isst. Bei einer Schule mit 560 Schülern werden pro Schüler inklusive Onkeln, Tanten, Eltern und Schülern etwa 3 Besucher erwartet, also 1 680 Besucher. Dies ergibt einen maximalen Absatz von 3360 Würstchen, bei 0 DM Verkaufspreis.
- Es gibt einen Preis, bei dem keine Würstchen mehr verkauft werden. Kommerziell verkaufte Bratwürstchen werden bis zu 4 DM mit lohnendem Umsatz gehandelt. Der Null-Umsatz-Preis muss also deutlich darüber liegen. Darüber hinaus handelt es sich um einen Verkauf für einen guten Zweck, der Erlös kommt ja letztlich den Kindern der Käufer zugute (die Verwendung des Erlöses sollte in der Aufgabenstellung der jeweiligen Schulsituation angepasst werden). Hier wird der Null-Umsatz-Preis mit 8 DM angenommen.

Mit linearer Interpolation zwischen den beiden Annahmen kann jetzt eine erste Gewinnfunktion erstellt werden. Da es sich dabei um eine Parabel handelt, kann nach Behandlung quadratischer Funktionen das Problem bis hierher auch analytisch behandelt werden.

[3] Diese Aufgabe wurde von Monika Seiffert, Staatliches Studienseminar, und Dr. Wolfgang Löding, IfL, entwickelt.

	A	B	C	D	E
1	Einkaufs-preis	0,8			
2	Null-Preis-Umsatz	3360	Null-Gewinn-Preis	8	
3	Schrittweite	-42	ist Umsatz bei 8 DM	=B85	
4	Preis	Anzahl verkaufter Würstchen	Einnahmen	Ausgaben	Gewinn
5	0	=B2	=A5*B5	=A5*B$1	=C5-D5
6	=A5+0,1	=B5+B$3	=A6*B6	=B6*B$1	=C6-D6
7	=A6+0,1	=B6+B$3	=A7*B7	=B7*B$1	=C7-D7

Tabelle 33: Würstchenverkauf mit linearer Interpolation

In A85 steht in der Tabelle 8 DM, über die Schrittweite in B3 muss die Anzahl der verkauften Würstchen so angepasst werden, dass in B85 0 steht. Dazu wird wie auch schon in anderen Aufgaben in D3 der Zielwert abgebildet, damit man gegebenenfalls durch Probieren die richtige Schrittweite erreicht.



Abb. 12: lineare Kostenfunktion für den Würstchenverkauf

Bei dieser Kostenfunktion liegt der maximale Gewinn bei 5 443,20 DM, wenn 1512 Würstchen für je 4,40 DM verkauft werden. Diese Kostenfunktion sollte jedoch kritisch betrachtet werden. Die hier dargestellten Verbesserungsvorschläge erheben keinen Anspruch auf Vollständigkeit:

- Für kleine Verkaufspreise wird der Umsatz zunächst kaum sinken.
- Für Verkaufspreise zwischen 7 DM und 8 DM sollte der Umsatz sich flacher an die Null annähern.
- Gegebenenfalls steigt der Umsatz zwischen 2 DM und 4 DM mit höherem Preis sogar an, da die Eltern wissen, dass sie mit dem Kauf einem guten Zweck dienen und ihr Geld für Angebote ausgeben, die der Cafeteria wirklich einen Gewinn verschaffen.

Diese und weitere Elemente können in der Tabellenkalkulation durch direktes Eingeben von Zahlen in Spalte B umgesetzt werden. Besser ist eine stückweise lineare Umsetzung, indem nur die Schrittweite in den Formeln in Spalte B jeweils geändert werden. In Tabelle 34 wurde durch das Einfügen einer eigenen Spalte gelöst, in der dann die jeweiligen Zuwachsraten eingegeben werden.

Für den Unterricht bietet es sich an, die erzielbaren Umsätze durch eine Umfrage unter den Schülerinnen und Schülern der Parallelklasse und der Schulelternschaft zu ermitteln. Dabei kann dann gleich ein wenig Statistik betrieben werden.

	A	B	C	D	E	F
1	Einkaufspreis		0,8			
2	Null-Preis-Umsatz		3360	Null-Gewinn-Preis	8	
3				ist Umsatz bei 8 DM	=C85	
4	Preis	Zuwachsrate	Anzahl verkaufter Würstchen	Einnahmen	Ausgaben	Gewinn
5	0		=C2	=A5*C5	=A5*C$1	=D5-E5
6	=A5+0,1	0	=C5+B6	=A6*C6	=C6*C$1	=D6-E6
7	=A6+0,1	0	=C6+B7	=A7*C7	=C7*C$1	=D7-E7

Tabelle 34: Würstchenverkauf mit nicht-linearer Interpolation

Die Umsatzfunktion wurde auf diese Weise folgendermaßen modelliert:



Abb. 13: Per Hand modellierte Umsatzfunktion für den Würstchenverkauf

Diese Modellierung ergibt einen maximalen Gewinn von 9 152 DM, wenn 2 080 Würstchen für je 5,20 DM verkauft werden.

In der Oberstufe könnte man diesen Verlauf grob durch ein Polynom dritten Grades beschreiben: $f(x) = 13{,}125x^3 - 157{,}5x^2 + 3360$, das mit Hilfe der Punkte (0 ; 3360) und (8 ; 0) sowie der Annahme, dass dort die Steigung Null ist, konstruiert wurde. Dann ergibt sich analytisch ein optimaler Preis von 3,74 DM mit einem Gewinn von 5 420 DM.

In dieser Aufgabe wird unter anderem sehr schön deutlich, wie ein Ergebnis einer mathematischen Untersuchung vom verwendeten Modell abhängt und dass Mathematik nicht immer eindeutige Ergebnisse liefert.

4.2.5 Extremwertaufgaben mit Kreisen

Kreisfläche und Kreisumfang können in Klasse 7 behandelt werden, wenn man auf Stufenflächenbetrachtungen verzichtet. Diese sind mangels des Satzes von Pythagoras nicht durchführbar.

In Klasse 7 steht die Kreiszahl in ihrer Eigenschaft als Proportionalitätsfaktor im Zentrum der Überlegung, wenn proportionale Zusammenhänge behandelt wurden. π kann dabei nur grob zwischen 2 und 4 liegend mit Hilfe der nebenstehenden Abbildung analytisch abgeschätzt werden.



Abb. 14: Abschätzung von π über den Flächeninhalt

Weitere Näherungen könnten durch Messungen (z.B. durch Auswiegen von ausgeschnittenen Kreisen und Quadraten) erreicht werden. Die ‘korrekte’ Größe für π wird dem Taschenrechner entnommen.

Liegt die Flächenformel vor, kann die Umfangsformel leicht mit Hilfe der folgenden Zeichnung (aus [8]) erschlossen werden, wobei die Überlegungen zur Proportionalität vorausgehen sollten.



Abb. 15: Von der Kreisfläche zum Kreisumfang

Alternativ kann man durch Abschätzungen des Umfanges und Messungen zunächst π über $u = 2\exists\pi\exists r$ bestimmen und dann über die Betrachtung in Abb. *15* den Umfang erarbeiten. Für die Abschätzung des Kreisumfanges hilft die folgende Abbildung.



Abb. 16: Abschätzung von π über den Umfang

Diese Überlegungen stehen im Unterricht erst nach der Erarbeitung der Proportionalität, also nicht direkt im Anschluss an die oben behandelten Aufgaben. Die folgenden Beispiele dienen dann gleichzeitig der Anwendung der Kreisformeln und der Wiederholung der vorangegangenen Extremwertaufgaben. Die Aufgaben sind bekannt, hier wird dennoch die Lösung mit der Tabellenkalkulation durchgeführt.

Aufgabe 19

Die klassische Aufgabe zur Optimierung des Blechverbrauches bei der Konservendose kann behandelt werden, wenn neben der Berechnung von Kreisfläche und Umfang auch die Volumenformel 'Grundfläche mal Höhe' bekannt ist.

	A	B	C	D	E	F	G
1	Volumen =	850	Weite =	0,1			
2	Radius	Grundfläche	Höhe	Mantelfläche	Oberfläche		
3	1	=A3^2*3,1415	=B$1/B3	=2*3,1415*A3*C3	=B3+D3		
4	=A3+D$1	=A4^2*3,1415	=B$1/B4	=2*3,1415*A4*C4	=B4+D4		
5	=A4+D$1	=A5^2*3,1415	=B$1/B5	=2*3,1415*A5*C5	=B5+D5		
6	=A5+D$1	=A6^2*3,1415	=B$1/B6	=2*3,1415*A6*C6	=B6+D6		
7	=A6+D$1	=A7^2*3,1415	=B$1/B7	=2*3,1415*A7*C7	=B7+D7		

Tabelle 35: Konservendose

Bei der hier verwendeten Genauigkeit liegt das Optimum bei einem Radius von 6,5cm. Die bekannte Erweiterung der Fragestellung, bei berücksichtigt wird, dass der Kreisdeckel aus einem Quadrat gestanzt wird, kann analog behandelt werden.

Aufgabe 20

In einem Dorf soll ein kombinierter Sport- und Festplatz gebaut werden. Dabei soll eine 400 m-Laufbahn entstehen, die den Rasenplatz umgibt. Der Rasenplatz soll die Form eines Rechteckes mit zwei angesetzten Halbkreisen erhalten. Der örtliche Sportverein und das Dorffestkomitee beauftragen jeweils einen Gutachter mit der Ermittlung der optimalen Form der Anlage. Die Gutachter kommen zu unterschiedlichen Ergebnissen. Welche Optimierung werden die verschiedenen Gruppen in Auftrag gegeben haben und welches sind deren Ergebnisse?

	A	B	C	D	E	F	G
1	Laufbahnlänge =	400	PI=	=PI()			
2	Schrittweite=	1					
3	Radius Kreis	Umfang Kreis	Länge Rechteck	Fläche Rechteck	Fläche Rasen		
4	0	=2*A4*D$1	=(B$1-B4)/2	=2*A4*C4	=D4+A4^2*D$1		
5	=A4+B$2	=2*A5*D$1	=(B$1-B5)/2	=2*A5*C5	=D5+A5^2*D$1		
6	=A5+B$2	=2*A6*D$1	=(B$1-B6)/2	=2*A6*C6	=D6+A6^2*D$1		
7	=A6+B$2	=2*A7*D$1	=(B$1-B7)/2	=2*A7*C7	=D7+A7^2*D$1		

Tabelle 36: Sportplatz mit Vorgabe Radius

Der Sportverein wünscht sich eine möglichst große Rechteckfläche für den Sportbetrieb. Die entsprechende Optimierung führt auf den üblichen Sportplatz. Das Festkomitee benötigt eine möglichst große Rasenfläche und erhält einen Kreis. Diese beiden Zielvorgaben sollten im Unterrichtsgespräch heraus gearbeitet werden. Die Rechteckfläche liefert eine Parabel.



Abb. 17: Fläche des Rechtecks beim Sportplatz

Das Optimum für die Rechteckfläche trifft man besser, wenn man in der Tabelle mit der Länge des Rechteckes beginnt. Diese Variationsmöglichkeit gibt es für die meisten Extremwertaufgaben. Zum Vergleich ist hier exemplarisch die zweite Variante mit dargestellt:

	A	B	C	D	E	F	G
1	Laufbahnlänge =	400	PI=	=PI()			
2	Schrittweite =	5					
3	Länge Rechteck	Umfang Kreis	Radius Kreis	Fläche Rechteck	Fläche Rasen		
4	0	=B$1-2*A4	=B4/(2*D$1)	=2*A4*C4	=D4+C4^2*D$1		
5	=A4+B$2	=B$1-2*A5	=B5/(2*D$1)	=2*A5*C5	=D5+C5^2*D$1		
6	=A5+B$2	=B$1-2*A6	=B6/(2*D$1)	=2*A6*C6	=D6+C6^2*D$1		
7	=A6+B$2	=B$1-2*A7	=B7/(2*D$1)	=2*A7*C7	=D7+C7^2*D$1		

Tabelle 37: Sportplatz mit Vorgabe Rechtecklänge

Aufgabe 21

Der Querschnitt eines großen Abwasserrohres besteht näherungsweise aus einem Halbkreis mit aufgesetztem Rechteck. Wie müssen die Abmessungen gewählt werden, um die Querschnittsfläche $0{,}7\,m^2$ mit möglichst geringem Materialaufwand zu realisieren?

	A	B	C	D	E	F	G
1	Fläche =	0,7	Weite =	0,05	PI=	=PI()	
2	Radius	F-Halbkreis	F-Rechteck	Höhe	U-Rechteck	U-Halbkreis	U-Gesamt
3	0,05	=A3^2*F$1/2	=B$1-B3	=C3/(2*A3)	=2*D3+2*A3	=F$1*A3	=E3+F3
4	=A3+D$1	=A4^2*F$1/2	=B$1-B4	=C4/(2*A4)	=2*D4+2*A4	=F$1*A4	=E4+F4
5	=A4+D$1	=A5^2*F$1/2	=B$1-B5	=C5/(2*A5)	=2*D5+2*A5	=F$1*A5	=E5+F5
6	=A5+D$1	=A6^2*F$1/2	=B$1-B6	=C6/(2*A6)	=2*D6+2*A6	=F$1*A6	=E6+F6
7	=A6+D$1	=A7^2*F$1/2	=B$1-B7	=C7/(2*A7)	=2*D7+2*A7	=F$1*A7	=E7+F7

Tabelle 38: Abwasserkanal

Es entsteht eine gebrochen rationale Funktion mit dem Minimum bei r = 45 cm. Dieser Radius ist in der Praxis zu groß, bei geringen Durchflussmengen sinken Strömungsgeschwindigkeit und Wassertiefe so weit ab, dass feste Bestandteile nicht mehr mitgenommen werden. Aus statischen Gründen ist es auch ungünstig, oben eine eckige Form zu wählen. Die im Tiefbau zu beobachtende Form legt es nahe, folgendermaßen zu modellieren.

Der untere Radius soll den festen Betrag 0,15m haben. Bei der Länge der Flanken tritt ein analoges Problem wie bei den Dachrinnen in Aufgabe 22 auf und kann genau so behandelt werden.



Abb. 18

	A	B	C	D	E	F	G
1	Fläche =	0,7	Weite =	0,05	PI=	=PI()	
2	Radius unten =	0,15					
3	dazu Fläche =	=B2^2*F1/2					
4	Radius oben	F-Oben	Fläche Trapez	Mittelwert Radien	Höhe Trapez	Kante Trapez	Umfang
5	0	=A5^2*F$1/2	=B$1-B$3-B5	=(B$2+A5)/2	=C5/D5	=WURZEL((B$2-A5)^2+E5^2)	=2*F5+F$1*(A5+B$2)
6	=A5+D$1	=A6^2*F$1/2	=B$1-B$3-B6	=(B$2+A6)/2	=C6/D6	=WURZEL((B$2-A6)^2+E6^2)	=2*F6+F$1*(A6+B$2)
7	=A6+D$1	=A7^2*F$1/2	=B$1-B$3-B7	=(B$2+A7)/2	=C7/D7	=WURZEL((B$2-A7)^2+E7^2)	=2*F7+F$1*(A7+B$2)

Tabelle 39: Abwasserkanal zweiter Ansatz

Das Optimum liegt bei 0,6 m oberen Radius mit einer Flankenlänge von nur 0,52 m. Diese Form entspricht nicht der Realität. Offensichtlich spielen statische und strömungstechnische Gesichtspunkte eine höhere Rolle als der Materialverbrauch.

Aufgabe 22

Eine Dachrinne wird aus einem 40 cm breiten Blechstreifen geschnitten. Wie ist das Fassungsvermögen der abgebildeten Dachrinnenformen (aus [7])? Für die verschiedenen Formen sind in Klasse 7 unterschiedliche Herangehensweisen erforderlich, da nicht alle Größen analytisch zur Verfügung stehen, sondern zeichnerisch ermittelt werden müssen. Die so hergestellten Wertetabellen können dann in die Tabellenkalkulation eingegeben werden, so dass die folgenden Rechnungen in gewohnter Weise darauf aufbauen können. Die Ergebnisse aus Dachrinnenform (4) können auf Form (5) übertragen werden.



Abb. 19: verschiedene Dachrinnenformen

	A	B	C	D	E	F	G
1	U=	40	PI=	=PI()			
2	Rund	mit Aufsatz					
3	Aufsatz	Radius	F-Aufsatz	F-Halbkreis	F-Gesamt		
4	0	=(B$1-2*A4)/D$1	=2*A4*B4	=B4^2*D$1/2	=C4+D4		
5	=A4+1	=(B$1-2*A5)/D$1	=2*A5*B5	=B5^2*D$1/2	=C5+D5		
6	=A5+1	=(B$1-2*A6)/D$1	=2*A6*B6	=B6^2*D$1/2	=C6+D6		
7	=A6+1	=(B$1-2*A7)/D$1	=2*A7*B7	=B7^2*D$1/2	=C7+D7		

Tabelle 40: Halbrunde Dachrinne mit geradem Aufsatz

Diese Dachrinnenform führt auf eine monoton fallende Funktion, da die halbrunde Dachrinne optimal ist. Für die Halbkreisform ergibt sich eine Querschnittsfläche von 254,65 cm^2.

	A	B	C	D	E	F	G
1	U=	40	Pi=	=PI()			
2	Rechteck			V-Form			
3	Höhe	Breite	Fläche	Alpha	Höhe	Breite	Fläche
4	0	=B$1-2*A4	=A4*B4	0	=SIN(D4*D$1/180)* B$1/2	=COS(D4*D$1/180) *B$1	=E4*F4/2
5	=A4+1	=B$1-2*A5	=A5*B5	=D4+5	=SIN(D5*D$1/180)* B$1/2	=COS(D5*D$1/180) *B$1	=E5*F5/2
6	=A5+1	=B$1-2*A6	=A6*B6	=D5+5	=SIN(D6*D$1/180)* B$1/2	=COS(D6*D$1/180) *B$1	=E6*F6/2
7	=A6+1	=B$1-2*A7	=A7*B7	=D6+5	=SIN(D7*D$1/180)* B$1/2	=COS(D7*D$1/180) *B$1	=E7*F7/2

Tabelle 41: Rechteckige Dachrinne und V-Form

Die Rechteckform entspricht Aufgabe 3. Der Querschnitt erreicht maximal 200,00 cm^2.

Für die V-Form sind hier für die Höhe und Breite die analytischen Ausdrücke angegeben. Die Funktionen cos und sin sind in den Tabellenkalkulationen für Bogenmaß definiert, so dass die Umrechnung erforderlich ist. Die Inhalte der Spalten E und F sind in den unteren Klassen zeichnerisch zu ermitteln. Als maximaler Querschnitt wird ebenfalls 200,00 cm^2 bestimmt.

	A	B	C	D	E	F
1	U=	40	Pi=	=PI()		
2	Wannen-Form					
3		V-Form für U=	=2*B1/3	s=	=B1/3	
4	Alpha	Höhe	Breite	Fläche	Recht-eckfläche	Gesamt
5	0	=SIN(A5*D$1/180)*C$3/2	=COS(A5*D$1/180)*C$3	=B5*C5/2	=E$3*B5	=D5+E5
6	=A5+5	=SIN(A6*D$1/180)*C$3/2	=COS(A6*D$1/180)*C$3	=B6*C6/2	=E$3*B6	=D6+E6
7	=A6+5	=SIN(A7*D$1/180)*C$3/2	=COS(A7*D$1/180)*C$3	=B7*C7/2	=E$3*B7	=D7+E7

Tabelle 42: Dachrinne in Wannenform

Die Wannenform kann als V-Form plus Rechteck aufgefasst werden, so dass die Daten aus den Reihen B und C nicht neu bestimmt werden müssen. Der maximale Querschnitt beträgt 231 cm^2.

4.2.6 Verteilungsprobleme

Diese Probleme führen naturgemäß auf umgekehrt proportionale Funktionen. Diesen Begriff kann man nach der Behandlung einiger Beispiele definieren und die Eigenschaften von umgekehrt proportionalen Funktionen thematisieren.

Aufgabe 23

Eine Klassenfahrt wird geplant. Für die Hin- und Rückfahrt (Hamburg - Lingen) macht ein Busunternehmen ein Pauschalangebot in Höhe von 2 200 DM. Die Bundesbahn bietet für die Gruppenreise einen Preis von 57 DM pro Person. Ab welcher Teilnehmerzahl ist die Busfahrt billiger?

	A	B	C	D	E	F	G
1	Kosten =	2200 DM	160 DM				
2	Personen		57 DM				
3		Busfahrt	Bahnfahrt				
4	1	=B$1/A4	=C$1/A4+C$2				
5	=A4+1	=B$1/A5	=C$1/A5+C$2				
6	=A5+1	=B$1/A6	=C$1/A6+C$2				
7	=A6+1	=B$1/A7	=C$1/A7+C$2				

Tabelle 43: Kostenverteilung Klassenreise mit Bustransfer

Erst ab 39 Schülern wird die Busfahrt billiger als die Bahnfahrt.

Die Fragestellung kann erweitert werden: Die Jugendherberge Lingen liegt 5 km vom Bahnhof entfernt. Die Busfahrt Bahnhof-Jugendherberge kostet bei einem ortsansässigen Busunternehmen pauschal 80 DM. Wie ändert sich dadurch die Kalkulation?
Über Spalte C ermittelt man, dass hier bei 35 Schülern die Kosten etwa gleich liegen. Es bietet sich an, die reine Kostenfrage in Zweifel zu ziehen: je nach Streckenauswahl muss man bei der Bahnfahrt bis zu dreimal umsteigen, wobei teilweise unter 10 Minuten Zeit vorgesehen sind, um von einem Bahnsteig zum anderen zu kommen, was bei einer Gruppenreise und möglichen Zugverspätungen zum Abenteuer geraten kann.

Aufgabe 24

Eine Schülerin möchte zu ihrem Geburtstag ihren 23 Klassenkameraden jeweils eine Süßigkeit oder etwas Ähnliches ausgeben. Sie hat dafür 30 DM zur Verfügung. Bevor sie in den Laden geht überlegt sie, ob sie lieber mehrere Kleinigkeiten für jeden kauft oder jeweils nur ein Stück für jeden und dafür etwas Größeres. Damit sie im Laden nicht rechnen muss, legt sie sich eine Tabelle an, in der die Zahl der Teile pro Schüler bei verschiedenen Preisen steht.

	A	B	C	D	E	F	G
1	Gesamt =	30 DM	Weite =	0,05			
2	Schülerzahl =	23					
3	Stückpreis	Anzahl	Stück pro Schüler				
4	0,05	=B$1/A4	=B4/B$2				
5	=A4+D$1	=B$1/A5	=B5/B$2				
6	=A5+D$1	=B$1/A6	=B6/B$2				
7	=A6+D$1	=B$1/A7	=B7/B$2				

Tabelle 44: Geburtstagskosten

Aufgabe 25

Zu einer großen Feier werden 13 Torten gebacken. Der Gastgeber grübelt, ob dies wohl reichen wird. Da er nicht weiß, ob alle Gäste kommen, verschafft er sich mit einer Tabelle einen Überblick. Er hat die Möglichkeit, die Torten in 8, 12 oder 16 Stücke aufzuschneiden.

	A	B	C	D	E	F	G
1	Anzahl =	13					
2	Zuschnitt	8	12	16			
3	Stücke =	=$B1*B2	=$B1*C2	=$B1*D2			
4	Gäste	Stücke p.P.	Stücke p.P.	Stücke p.P.			
5	30	=B$3/$A5	=C$3/$A5	=D$3/$A5			
6	=A5+1	=B$3/$A6	=C$3/$A6	=D$3/$A6			
7	=A6+1	=B$3/$A7	=C$3/$A7	=D$3/$A7			

Tabelle 45: Kuchenverteilung

Wie sind hier Bruchteile von Kuchenstücken zu werten? Die drei Graphen können verglichen werden. Was passiert, wenn 'halbe‘ Gäste kommen?

4.2.7 Wachstumsprozesse

Typische Wachstumsprozesse verlaufen im ersten Ansatz exponentiell. Bei Behandlung der folgenden Beispiele in Stufe 7 sollten keine Funktionsterme behandelt werden, aber die Qualität des Funktionsgraphen diskutiert werden.

Aufgabe 26

Ein Schüler erhält zur Einschulung ins Gymnasium von seiner Oma 1 200 DM auf ein Sparbuch eingezahlt. Er erhält jährlich 4% Zinsen. Wie viel Geld hat er zum Abitur? Wie viel Geld wäre es, wen er das Geschenk schon zur Geburt erhalten hätte?

	A	B	C	D	E	F	G
1	Kapital =	1200					
2	Zinssatz	0,04					
3	Jahre	Kapital					
4	0	=B1					
5	=A4+1	=B4*(1+B$2)					
6	=A5+1	=B5*(1+B$2)					
7	=A6+1	=B6*(1+B$2)					

Tabelle 46: Zinseszinsentwicklung

Die Eingabe 0,04 für den Zinssatz ist in der Tabellenkalkulation zwingend, wenn man die %-Formatierung nutzen will. Im Unterricht sollte die Umsetzung 4 % = 0,04 für Prozentzahlen für die Schülerinnen und Schüler zum Standard werden, da dadurch die 100 aus der Prozentrechnung herausfällt und die ganze Prozentrechnung deutlich einfacher wird. So werden 16% Mehrwertsteuer aufgeschlagen, indem mit 1,16 multipliziert wird. Verfährt man auf diese Weise, benötigt man keine umständliche Formeln und kann schnell mit dem Taschenrechner Ergebnisse ausrechnen.

Aufgabe 27

Auch die bekannte Schachbrettaufgabe kann mit der Tabellenkalkulation durchgerechnet werden:

	A	B	C	D	E	F	G
1			Masse pro 1000 Körner =	0,02 kg			
2	Feld	Reiskörner auf dem Feld	Summe	Gewicht			
3	1	1	=B3	=C3/1000*D$1			
4	=A3+1	=B3*2	=C3+B4	=C4/1000*D$1			
5	=A4+1	=B4*2	=C4+B5	=C5/1000*D$1			
6	=A5+1	=B5*2	=C5+B6	=C6/1000*D$1			
7	=A6+1	=B6*2	=C6+B7	=C7/1000*D$1			

Tabelle 47: Schachbrett und Reiskörner

Die Berechnung ergibt eine Summe von 368 934 881 474 t, bei ca. 10g auf 500 Reiskörner. Diesen Wert kann man genauer ermitteln: Wenn jeder 1 000 Körner abzählt, erhält man eine besser wägbare Masse und kann gleich etwas Statistik betreiben. Die Welternte 1991 betrug etwa 510 000 000t Reis. Die Schülerinnen und Schüler sollten den Zusammenhang zwischen den Spalten B und C erkennen, der den Zugang zur analytischen Bestimmung erleichtert.

Aufgabe 28

Der Wettlauf von Achilles mit der Schildkröte über die Tabellenkalkulation liefert zunächst das Anschauungsmaterial für die Idee der Endlichkeit der geometrischen Reihe.

	A	B	C	D	E	F	G
1	Weg Achilles	Weg Schildkröte	Standort Schildkröte				
2	0	10	=B2				
3	=B2	=A3/10	=C2+B3				
4	=B3	=A4/10	=C3+B4				
5	=B4	=A5/10	=C4+B5				
6	=B5	=A6/10	=C5+B6				
7	=B6	=A7/10	=C6+B7				

Tabelle 48: Achilles und die Schildkröte

Diese Aufgabe erfordert natürlich keine Tabellenkalkulation. Die Zahlenkolonne macht die analytischen Betrachtungen jedoch viel einfacher. Für die Zielrichtung ‚Funktionsbegriff' steht der Graph im Vordergrund der Betrachtungen.

Aufgabe 29

Der Holzbestand eines jungen Waldes wächst jedes Jahr um den gleichen Faktor. Innerhalb von 12 Jahren nimmt er um ca. 50 % zu. Auf einem Hektar Waldboden werden etwa 3 m^3 Sprösslinge gepflanzt. Wie entwickelt sich der Wald in den ersten 20 Jahren? (aus [8])

	A	B	C	D	E	F	G
1	Faktor	1,034366	Soll nach 12 Jahren	=1,5*B3			
2	Jahr	Menge	Ist nach 12 Jahren	=B15			
3	0	3					
4	=A3+1	=B3*B$1					
5	=A4+1	=B4*B$1					
6	=A5+1	=B5*B$1					
7	=A6+1	=B6*B$1					

Tabelle 49: Entwicklung des Holzbestandes

Hier muss über Probieren der Faktor in B1 gefunden werden, wie schon in den Aufgaben zur linearen Interpolation. Die Startmenge 3 m^3 Sprösslinge pro Hektar kann als Abschätzungsaufgabe diskutiert werden: Wie dick und lang ist ein Schössling, wie viele werden auf einem Hektar etwa gepflanzt?

Aufgabe 30

Ein Auto zum Neupreis von 24 000 DM verliert innerhalb eines Jahres ein Viertel seines Wertes (bezogen auf den Beginn des Jahres). Wie entwickelt sich der Wert des Autos im Laufe der Zeit?

Man spart jedes Jahr 3 000 DM und erhält jeweils am Ende des Jahres 4 % Zinsen. Nach wie viel Jahren kann man sich wieder ein Auto für 24 000 DM kaufen, wenn man den alten Wagen in

Zahlung gibt? Wie verändert sich die Situation, wenn man die Preissteigerung der Neuwagen mit berücksichtigt? (nach [8])

	A	B	C	D	E	F	G
1	Wert =	24000					
2	Sparbetrag =	3000	Zinsen =	0,04		Inflation =	0,02
3	Jahr	Wert	Einzah-lung	Konto-stand	Zinsen	mit Rest-wert	Kosten Wagen
4	0	=B1				=B4+D4	=B4
5	=A4+1	=B4*0,75	=B$2	=C5+D4 +E4	=D$2*D5	=B5+D5	=G4*(1+G$2)
6	=A5+1	=B5*0,75	=B$2	=C6+D5 +E5	=D$2*D6	=B6+D6	=G5*(1+G$2)
7	=A6+1	=B6*0,75	=B$2	=C7+D6 +E6	=D$2*D7	=B7+D7	=G6*(1+G$2)

Tabelle 50: Autofinanzierung

Diese komplexe Aufgabe sollte man zwar als Ganzes stellen, den Wertverfall und die Refinanzierung jedoch in der Bearbeitung nacheinander durchführen lassen. Die Fragestellung kann auch umformuliert werden: Wie viel muss man jährlich sparen, um nach vier Jahren ein neues Auto kaufen zu können (ca. 4 300 DM)? Wie hängt dies vom Anschaffungspreis ab?

Aufgabe 31

Wanderratten sind ausgesprochene Allesfresser. Sie finden oft günstige Lebensbedingungen vor und vermehren sich dann stark: Angenommen, eine Population von 20 Wanderratten vermehrt sich täglich um 1,5%. Wie groß ist die Population nach einer Woche, einem Monat, einem Jahr? (aus [8]).

	A	B	C	D	E	F	G
1	Startpopulation	20					
2	Wachstum	0,015					
3	Tag	Population					
4	0	=B1					
5	=A4+1	=B4*(1+B$2)					
6	=A5+1	=B5*(1+B$2)					
7	=A6+1	=B6*(1+B$2)					

Tabelle 51: Wanderratten

Nach einer Woche sind es 22 Ratten, nach 30 Tagen 31 Ratten und nach einem Jahr 4008 Ratten.

Aufgabe 32

Auf einem abgegrenzten Gebiet (eine größere Wiese mit umliegenden Abgrenzungen wie Flüssen, Steilhängen, Autobahn) leben Kaninchen. Eine Zählung hat ergeben, dass es 550 Tiere sind. Man möchte prognostizieren, wie viele Kaninchen im folgenden und in den darauf folgenden Jahren auf der Wiese leben werden, um die Kosten für weitere Zählungen gering zu halten.
Man sammelt verschiedene Einflussfaktoren: Geburtenrate (8 Kaninchen pro Wurf, bei zwei Würfen pro Jahr pro Kaninchenpärchen) und verschiedene Todesursachen (Fuchs, Autoverkehr, Krankheiten etc.). Summiert man diese Faktoren, so ergibt sich ein einfaches Kaninchenfortpflanzungsgesetz dergestalt, dass sich die Zahl der Kaninchen in jedem Jahr verdreifacht (oder ein ähnlicher Faktor).

	A	B	C	D	E	F	G
1	Fortpflanzungsfaktor =	3					
2	Startbevölkerung =	500					
3	Jahr	Kaninchen					
4	0	=B$2					
5	=A4+1	=B$1*B4					
6	=A5+1	=B$1*B5					
7	=A6+1	=B$1*B6					

Tabelle 52: proportionale Kaninchenfortpflanzung

Das betrachtete Modell ist nicht sinnvoll, die Anzahl der Kaninchen wächst in kurzer Zeit so stark, dass die Wiese übervölkert wird. Zunächst wird man versuchen, den Fortpflanzungsfaktor zu verändern, wobei keine qualitative Änderungen eintreten, es sei denn, der Faktor liegt unter eins.
Eine weitere Veränderung wäre, in jedem Jahr eine konstante Menge Kaninchen zu entnehmen (z.B. durch Abschuss oder andere konstant wirkende Faktoren), in der Hoffnung, dass dadurch die Kaninchenpopulation beschränkt bleibt.

	A	B	C	D	E	F	G
1	Fortpflanzungsfaktor =	1,1	Abschuss =	60			
2	Startbevölkerung =	500					
3	Jahr	Kaninchen					
4	0	=B$2					
5	=A4+1	=B$1*B4-D$1					
6	=A5+1	=B$1*B5-D$1					
7	=A6+1	=B$1*B6-D$1					

Tabelle 53: lineare Kaninchenfortpflanzung

Hier gibt es qualitativ zwei mögliche Entwicklungen: Bei den in der Tabelle angenommenen Werten stirbt die Population aus, erhöht man hier den Fortpflanzungsfaktor auf 1,2, wächst die Zahl der Kaninchen wieder über alle Grenzen[4].

[4] Vollständige Analysen zu diesem und den folgenden Phänomenen findet man in [5]

Experimentiert man mit B1, so ergibt sich bei einem Fortpflanzungsfaktor zwischen Null und eins und einem negativen Abschuss unabhängig von der Startbevölkerung eine stabile Kaninchensituation.

In dem Wachstumsgesetz muss für eine sinnvolle Populationsentwicklung ein begrenzender Term enthalten sein, der die beschränkten natürlichen Ressourcen modelliert. Dieser Term muss zwischen Null und eins liegen, wobei er für große Populationen nahe bei Null und für kleine nahe bei eins sein muss.

Ein möglicher Faktor, der dies leistet, ist (MAX-x)/MAX, wobei MAX die maximal mögliche Population in dem Biotop ist und das neue Wachstumsgesetz lautet dann 3x(MAX-x)/MAX.

	A	B	C	D
1	Fortpflanzungsfaktor =	2,3		
2	Startbevölkerung =	300	Maximale Bevölkerung =	2000
3	Jahr	Kaninchen		
4	0	=B$2		
5	=A4+1	=B$1*B4*(D$2-B4)/D$2		
6	=A5+1	=B$1*B5*(D$2-B5)/D$2		
7	=A6+1	=B$1*B6*(D$2-B6)/D$2		

Tabelle 54: logistische Kaninchenfortpflanzung

Für positive Fortpflanzungsfaktoren kleiner als drei ergibt sich eine Stabilisierung der Kaninchenpopulation. Für Werte zwischen drei und vier ergeben sich Schwingungszustände, die auch in [5] behandelt werden. Im Zusammenhang mit geometrischer Iteration eignet sich diese Aufgabensequenz gut zur sinnhaften Einführung von quadratischen Funktionen.



Abb. 20: logistische Kaninchenfortpflanzung

Für die Einführung des Funktionsbegriffes sollten die auftretenden Funktionen beschrieben werden: das Kaninchenfortpflanzungsgesetz und dessen Resultat, die Kaninchenpopulationsentwicklung.

In diesem Zusammenhang wird häufig eine andere Darstellungsform gewählt, das Spinnwebverfahren:

	A	B	C	D	E
1	Faktor		3,23	Start	0,042
2	x	Winkelhalbierende	f(x)	Iteration	
3	0	=A3	=C$1*A3*(1-A3)	=E1	0
4	=A3+0,01	=A4	=C$1*A4*(1-A4)	=D3	=C$1*D4*(1-D4)
5	=A4+0,01	=A5	=C$1*A5*(1-A5)	=E4	=D5
6	=A5+0,01	=A6	=C$1*A6*(1-A6)	=D5	=C$1*D6*(1-D6)
7	=A6+0,01	=A7	=C$1*A7*(1-A7)	=E6	=D7

Tabelle 55: logistische Kaninchenfortpflanzung mit Spinnwebverfahren

In Spalte B werden die Daten für die Winkelhalbierende g(x) = x erzeugt, in Spalte C die Werte des Fortpflanzungsgesetzes. Aus diese wurde in eine Zeichnung hergestellt. Danach wurde die Iteration in den Spalten D und E erzeugt, so dass jeweils zwei Linienstücke entstehen: ‚senkrecht zur Funktion‘ und ‚waagerecht zur Winkelhalbierenden‘[5]. Die Zeichnung aus diesen Daten wird dann der anderen Zeichnung hinzugefügt.



Abb. 21: Spinnwebverfahren für logistisches Kaninchenwachstum mit Steuergrafiken

Hier ist es besonders schön, für den Startwert und den Faktor in C1 eine separate Grafik zu erstellen und über das Verschieben der Punkte in diesen Grafiken das Spinnwebverfahren zu steuern.

Berücksichtigt man, dass die Kaninchen nicht sofort geschlechtsreif sind, gelangt man zu Fibunacci-Folgen: Hier wird jedes Kaninchenpaar nach Ablauf einer Fortpflanzungsperiode pro Periode zwei Kaninchen zeugen.

[5] Die Erläuterung zum Kopieren den alternierenden Zellen steht in Tabelle 5.

	A	B	C	D	E	F	G
1	Elternpaar	Kinderpaar	gesamt				
2	1	0	=A2+B2				
3	=C2	=A2	=A3+B3				
4	=C3	=A3	=A4+B4				
5	=C4	=A4	=A5+B5				
6	=C5	=A5	=A6+B6				
7	=C6	=A6	=A7+B7				

Tabelle 56: Fibunacci-Folge

Die Untersuchungen zur Fibunacci-Folge sollen hier nicht ausgebreitet werden, man kann sie beispielsweise aus [7] entnehmen. Die Tabellenkalkulation kann dabei für das Entdecken von Zusammenhängen nützlich sein, die Analyse ist jedoch in der Oberstufe anzusiedeln.

4.2.8 Lineare und stückweise lineare Zusammenhänge

Aufgabe 33

Seit dem 1.1.1998 gelten in Hamburg neue Gastarife in drei Preisstufen:

Preisstufe	Jahresabnahme in kWh	Grundpreis in DM/Jahr	Arbeitspreis in Pf/kWh
I	bis 11.000	172,50	5,4740
II	11.000 – 60.000	239,20	4,8645
II	über 60.000	322,00	4,7265

Tabelle 57: Gastarife in Hamburg

Sind die Übergänge zwischen den Tarifen stetig (also frei von Sprüngen)? Welche Auswirkungen haben die verschiedenen Tarife in einer grafischen Darstellung?

	A	B	C	D	E
1		Stufe 1	Stufe2	Stufe 3	
2	Grundpreis	172,5	239,2	322	
3	Arbeitspreis	=5,474/100	=4,8645/100	=4,7265/100	
4	Menge	Kosten	Kosten	Kosten	gültiger Tarif
5	0	=B$2+A5*B$3	=C$2+A5*C$3	=D$2+A5*D$3	=WENN(A5<11000;B5; WENN(A5<60000;C5;D5))
6	=A5+1000	=B$2+A6*B$3	=C$2+A6*C$3	=D$2+A6*D$3	=WENN(A6<11000;B6; WENN(A6<60000;C6;D6))
7	=A6+1000	=B$2+A7*B$3	=C$2+A7*C$3	=D$2+A7*D$3	=WENN(A7<11000;B7; WENN(A7<60000;C7;D7))

Tabelle 58: Gastarife in Hamburg

Die Zahlen zeigen einen kleinen Sprung in Höhe von 34 Pf bei 11 000 kWh, bei 60 000 kWh gibt es keinen Sprung. Die verschiedenen Steigungen erkennt man nicht, wenn man nur den gültigen Tarif darstellt. Daher sind hier die drei Spalten B, C, und D vollständig gezeichnet. Will man den gültigen Tarif zeichnen, empfiehlt es sich, diese Spalte direkt neben der Menge anzuordnen.



Abb. 22: Gastarife in Hamburg

Aufgabe 34

Die gültigen Telefontarife sind sehr unübersichtlich. Für den Vergleich der Gesprächskosten muss man sowohl Wochenenden von Arbeitstagen unterscheiden als auch die Tageszeit berücksichtigen. Dann gibt es immer noch eine Vielzahl von Anbietern mit unterschiedlichen Preisen und Taktlängen. An dieser Stelle werden exemplarisch vier Tarife für das Zeitfenster Montag bis Freitag, 18.00 bis 21.00 Uhr, verglichen.

	A	B	C	D	E	F	G	H	I
1		Mo- Fr	von	18.00	bis	21.00			
2		Analog		ISDN		Aktiv Plus		o.tel.o	
3		Taktlänge in Sekunden	Takt-preis	Taktlänge in Sekunden	Takt-preis	Taktlänge in Sekunden	Takt-preis	Taktlänge in Sekunden	Takt-preis
4	City	150	0,12	240	0,12	60	0,03	1	0,000 833
5									
6	Zeit	Anzahl Takte	Ent-geld	Anzahl Takte	Ent-geld	Anzahl Takte	Ent-geld	Anzahl Takte	Ent-geld
7			=B3		=D3		=F3		=H3
8	0	=AUFRUNDEN($A9/B$5;0)	=B9*C$5	=AUFRUNDEN($A9/D$5;0)	=D9*E$5	=AUFRUNDEN($A9/F$5;0)	=F9*G$5	=AUFRUNDEN($A9/H$5;0)	=H9*I$5
9	=A9+10	=AUFRUNDEN($A10/B$5;0)	=B10*C$5	=AUFRUNDEN($A10/D$5;0)	=D10*E$5	=AUFRUNDEN($A10/F$5;0)	=F10*G$5	=AUFRUNDEN($A10/H$5;0)	=H10*I$5

Tabelle 59: Telefontarife

Interessant ist an dieser Stelle die Auswirkung unterschiedlicher Taktzeiten. Über das Internet können die benötigten Daten recht einfach auch vollständig recherchiert werden, so dass deutlich umfangreichere Vergleiche hergestellt werden können.

Die Grafik beantwortet für jeden Betrag die Frage eindeutig (und somit falsch), wie lange man für ihn telefonieren kann. Sprünge werden nicht darstellt, weil die Tabellenkalkulation die Linien durchzieht. Hier kann man mit den Schülerinnen und Schülern die Grenzen so einer numerischen Darstellung diskutieren.



Abb. 23: Telefontarife

Aufgabe 35

Vor dem Kauf eines Mobilfunktelefons sollen die Kosten des Telefons ermittelt werden. Dazu werden von verschiedenen Anbietern die Tarife eingeholt. Die Tarife der verschiedenen Anbieter haben sich in den letzten Jahren einander stark angenähert, daher werden hier nur je ein Tarif mit Grundgebühr und ein Prepaidtarif verglichen.

	A	B	C	D	E
1			Tarif mit Vertragsbindung		Prepaidtarif
2	monatlicher Grundpreis		24,95		0
3	Grundpreis über 2 Jahre		=C2*24		=E2*24
4	Anschaffung Telefon		1		100
5	Anmeldegebühren		0		0
6	Summe Fixkosten		=C3+C4+C5		=E3+E4+E5
7	Fixkosten pro Monat		=C6/24		=E6/24
8	monatliche Aufladung				=50/12
9	Sunshine	0,3	0,99	0,3	1,69
10	Moonshine	0,2	0,39	0,2	0,69
11	Sun local	0,25	0,15	0,2	0,39
12	Moon local	=100%-B11-B10-B9	0,15	0,2	0,15
13	weekend			=100%-D12-D11-D10-D9	0,15
14	Minutenpreis (Mittel)		=C9*B9+C10*B10+C11*B11+C12*B12		=E9*D9+E10*D10+E11*D11+E12*D12+D13*E13
15	monatliche Telefonzeit		Tarif mit Vertragsbindung		Prepaidtarif
16	0		=C$7+$A16*C$14		=WENN($A16*E$14<E$8; E$7+E$8;$A16*E$14+E$7)
17	=A16+1		=C$7+$A17*C$14		=WENN($A17*E$14<E$8; E$7+E$8;$A17*E$14+E$7)

Tabelle 60: Tarife für Mobiltelefone

Die Modellierung dieses Vergleiches ist eine Lernsituation, die über einen längeren Zeitraum des Mathematikunterrichts trägt und dabei verschiedene mathematische Fähigkeiten erfordert. Diese Aufgabe kann mit dem Zusammentragen der verschiedenen Tarifinformationen zu einem kleinen Mathematikprojekt entwickelt werden. Es sollten nach Möglichkeit aktuelle Prospekte der Anbieter oder Angebote aus Internetseiten verwendet werden. Wesentliche Tätigkeiten der Datensichtung und Auswahl der notwendigen Informationen werden von den Schülerinnen und Schülern erledigt.

Beim Prepaidtarif ermitteln sich die laufenden Kosten aus dem Maximum der beiden Zahlen ‚tatsächlich zu zahlende Gebühren' und ‚Mindestaufladung'. Dies leistet die MAX()-Funktion der Tabellenkalkulation oder die wenn()-Funktion, die hier verwendet wurde. Ein wichtiger Faktor sind die Kosten des Telefons: die Netzbetreiber subventionieren (zum Zeitpunkt der Drucklegung dieser Handreichung noch) die Anschaffung eines Telefons in unterschiedlicher Höhe, je nachdem, ob ein Prepaidtarif oder ein Tarif mit Grundgebühr gewählt wird. Dabei sind die Handykosten jedoch auch bei den Prepaidtarifen sehr stark gesunken. Die Minutenpreise zum Festnetz wurden als Mittelwert zwischen den vier in den Prospekten auftretenden Minutenpreisen modelliert: tagsüber und nachts jeweils zur Wunschvorwahl oder in den übrigen Teil des Netzes. Dabei wurden für die Nutzungen unterschiedliche Prozentanteile angenommen, die das Profil des potentiellen Nutzers darstellen. Dies kann noch ergänzt werden, wenn man die Tarife zu anderen Mobilfunktelefonen mit berücksichtigt. Hier kann man beispielsweise diskutieren, ob man Bekannte berücksichtigt, die man über ein Funktelefon erreichen kann. Weiterhin sind nicht nur die Kosten sondern auch die Qualität der Leistungen relevant. Die Netze haben eine unterschiedliche Ausbaudichte und Empfangsqualität. Die mathematische Modellbildung wird hier durch eine Reihe von individuellen Gesichtspunkten beeinflusst, woran deutlich werden kann, wie mathematisch errechnete ‚Wahrheiten' vom Modell abhängen. Weitere Untersuchungen könnte man für die Profi-Tarife anstellen, die höhere Grundgebühren bei niedrigeren Gesprächsgebühren vorsehen. Anschließen kann sich die analytische Betrachtung der Frage, ab welcher Telefonierzeit welcher Tarif für ein vorgegebenes Nutzerprofil am günstigsten ist. Für die oben durchgeführten Berechnungen ergibt sich die folgende Grafik:



Abb. 24: Tarife für Mobiltelefone

Man sieht, dass zum Zeitpunkt der Drucklegung die grundgebührenfreien Tarife für relativ weite Bereiche in der Nutzungszeit pro Monat (x-Achse, in Minuten) günstiger sind als die Tarife mit Grundgebühren

Aufgabe 36

Würfel sollen im Werkunterricht als Leistengerüst gebaut werden und anschließend mit Stoff bespannt werden. Wie entwickelt sich der Materialverbrauch in Abhängigkeit von der Kantenlänge? Die Leisten kosten pro Meter 1,50 DM, der Stoff pro laufendem Meter 7 DM, bei einer Breite von 1,4 m. Für die 28 Schüler einer Klasse stehen 400 DM für Material zur Verfügung. Wie groß können die Schüler die Würfel bauen?

	A	B	C	D	E	F	G
1	Leiste Preis pro m	1,5		Stoffpreis pro m	35		
2				Stoffbreite	1,4		
3				Stoff Preis pro m^2	=E1/E2		
4	Kantenlänge	Leisten-länge	Leisten-preis	Stofffläche	Stoffpreis	Gesamtpreis 28 Schüler	
5	0,05	=A5*12	=B$1*B5	=6*A5^2	=D5*E$2	=C5+E5	
6	=A5+0,05	=A6*12	=B$1*B6	=6*A6^2	=D6*E$2	=C6+E6	
7	=A6+0,05	=A7*12	=B$1*B7	=6*A7^2	=D7*E$2	=C7+E7	

Tabelle 61: Würfelbau

Bei den angegebenen Verhältnissen kann jede Schülerin bzw. jeder Schüler einen Würfel von 45 cm Kantenlänge bauen. Die Kosten für die Leisten sind eine proportionale Funktion, die Kosten für den Stoff eine Parabel.

Aufgabe 37

Ein Parkhaus erhebt die folgenden Gebühren: Die 1. Stunde ist frei, jede weitere Stunde kostet 2 DM, wobei auf die Minute genau abgerechnet wird. Die Parkgebühren sollen als Liste und als Grafik ausgehängt werden.

	A	B	C	D	E	F	G
1	Stundenpreis	2	Minutenpreis	=B1/60			
2	Kostenfreie Zeit	60					
3	Zeit in min	Kosten					
4	=B2	=(A4-B$2)*D$1					
5	=A4+1	=(A5-B$2)*D$1					
6	=A5+1	=(A6-B$2)*D$1					
7	=A6+1	=(A7-B$2)*D$1					

Tabelle 62: Parkgebühren

Aufgabe 38

In der Zeitung steht ein Angebot: 25 kg Kartoffeln kosten 40 DM. Um dies mit den Preisen anderer Geschäfte vergleichen zu können, die andere Mengen anbieten, erstellt Fritz eine Tabelle, in der steht, wie viel 1 kg, 2 kg, ... kosten würden.

	A	B	C	D	E	F	G
1	Kartoffelmenge	25					
2	Preis für die Menge	40					
3	Preis pro kg	=B2/B1					
4	Menge	Preis					
5	1	=A5*B$3					
6	=A5+1	=A6*B$3					
7	=A6+1	=A7*B$3					

Tabelle 63: Preistabelle

Aufgabe 39

Ein ICE fährt um 11:05 Uhr ab Hamburg (Kilometerstein 0) nach Hannover (Kilometerstein 183) und kommt um 12:25 Uhr an. Um 10:57 Uhr fährt ein ICE von Hannover nach Hamburg und erreicht Hamburg um 12:28 Uhr.
Bei welchem Kilometerstein treffen sich die beiden Züge, vorausgesetzt, sie fahren mit konstanter Geschwindigkeit und verwenden dieselbe Strecke?[6]

	A	B	C
1		Hannover - Hamburg	Hamburg - Hannover
2	Abfahrt	10:57	11:03
3	Ankunft	12:28	12:25
4	Fahrzeit in Stunden	=B3-B2	=C3-C2
5	Fahrzeit dezimal in Stunden	=STUNDE(B4)+MINUTE(B4)/60	=STUNDE(C4)+MINUTE(C4)/60
6	Start	183	0
7	Ziel	0	183
8	Strecke	=B7-B6	=C7-C6
9	Geschwindigkeit in km/h	=B8/B5	=C8/C5
10	Geschwindigkeit in km/min	=B9/60	=C9/60
11	Uhrzeit	Hannover - Hamburg	Hamburg - Hannover
12	=b2	=B6	
13	=A12+"00:01:00"	=B12+B$10	
14	=A13+"00:01:00"	=B13+B$10	
15	=A14+"00:01:00"	=B14+B$10	
16	=A15+"00:01:00"	=B15+B$10	
17	=A16+"00:01:00"	=B16+B$10	
18	=A17+"00:01:00"	=B17+B$10	=C6
19	=A18+"00:01:00"	=B18+B$10	=C18+C$10
20	=A19+"00:01:00"	=B19+B$10	=C19+C$10
21	=A20+"00:01:00"	=B20+B$10	=C20+C$10

Tabelle 64: Bahnfahrt

6 Aktuelle Daten erhält man sehr komfortabel über http://www.bahn.de

Die meiste Arbeit an dieser Tabelle macht das Behandeln der Uhrzeiten. Will man dieses Problem umgehen, rechnet man alles in Minuten ab 10:57 Uhr. In Zelle A18 ist es 11:03.



Abb. 25: Bahnfahrt

4.2.9 Taschenrechner erkunden

Bei der Einführung des Taschenrechners stellen die Schülerinnen und Schüler regelmäßig die Frage nach Funktionen, die aufgrund noch nicht behandelter mathematischer Inhalte nicht beantwortet werden können. Einige Tasten wie die Wurzelfunktion können dabei über einschlägige Werte verständlich gemacht werden, sin, cos, e^x oder ln entziehen sich jedoch diesem Zugang. Unter dem Gesichtspunkt des Funktionsbegriffes ist es sinnvoll, sich unbekannten Funktionen dadurch zu nähern, dass man eine Wertetabelle und eine Zeichnung erstellt. Mit Hilfe einer Tabellenkalkulation ist dies einfach möglich.

Für die Winkelfunktionen sind dabei verschieden umfangreiche Behandlungen möglich. In Tabellenkalkulationen sind diese Funktionen generell in Bezug auf das Bogenmaß definiert. Verwendet man dieses, kann man in unteren Klassen keinen Bezug zu Winkeln herstellen. Man erhält dann den Bezug ‚sin beschreibt eine Schwingung', die als propädeutische Assoziation durchaus sinnvoll ist. Will man das Gradmaß benutzen, so muss man zusätzliche Funktionen definieren (hier im Format von Office 97):

```
Function sinus(alpha)
  sinus = Sin(1.74532925199433E-02 * alpha)
End Function
```

```
Function cosinus(alpha)
  cosinus = Cos(1.74532925199433E-02 * alpha)
End Function
```

Diese definierten Funktionen sind dann die Winkelfunktionen für das Gradmaß und können wie normale Funktionen in der Tabellenkalkulation verwendet werden. Man muss die Funktionen für die Behandlung im Unterricht vorbereiten und die Schülerinnen und Schüler dann die vorbereitete Datei verwenden lassen.

Stellt man dann eine Wertetabelle auf und erzeugt Grafiken, so kann man die folgenden Ergebnisse erzielen:

	A	B	C	D	E	F	G
1	alpha	sinus(alpha)	cosinus(alpha)				
2	0	=sinus(A2)	=cosinus(A2)				
3	=A2+1	=sinus(A3)	=cosinus(A3)				
4	=A3+1	=sinus(A4)	=cosinus(A4)				
5	=A4+1	=sinus(A5)	=cosinus(A5)				
6	=A5+1	=sinus(A6)	=cosinus(A6)				
7	=A6+1	=sinus(A7)	=cosinus(A7)				

Tabelle 65: Wertetabelle sin und cos



Abb. 26

Trägt man sin und cos nicht über der unabhängigen Variablen auf, sondern verwendet die Spalte B für die x-Achse, so erhält man den Cosinus als Funktion des Sinus.
Zumindest als Eindruck kann man den Schülerinnen und Schülern eine Assoziation von Winkelfunktion mit Kreis vermitteln. Pflegt man solche Assoziationen im Unterricht, entsteht im Laufe der Jahre ein Gedankennetzwerk, dass Schülerinnen und Schüler viel besser befähigt, neue mathematische Inhalte in ein Denksystem einzuordnen. Diese Pflege kann beispielsweise durch die langfristige Nutzung des eigenen Funktionenkatalogs geschehen, in den diese Grafiken natürlich mit hinein gehören.



Abb. 27: cosinus als Funktion von sinus

Analog können Logarithmus- und Exponentialfunktion gezeichnet werden. Für die Exponentialfunktionen können dabei Beziehungen zu den Wachstumsprozessen entdeckt werden. Zeichnet man Exponential- und Logarithmusfunktion in eine Zeichnung, wird für die Schülerinnen und Schüler aufgrund der Spiegelsymmetrie klar, dass diese Funktionen zusammengehören.

4.3 Leistungsüberprüfung

Wird die Tabellenkalkulation im Unterricht verwendet, sollten die Fähigkeiten zum Umgang mit dieser Software auch Gegenstand von Klassenarbeiten sein. Dabei sollte der Aspekt des Termverständnisses im Vordergrund stehen. Für den Aspekt der Modellierung eines mathematischen Sachverhaltes kann in der Klassenarbeit eine geeignete Aufgabe aus den hier vorgestellten verwendet werden oder leicht abgewandelt werden, wobei die Berechnungen dann mit dem Taschenrechner durchgeführt werden müssen. Für Klassenarbeiten am Computer sind die meisten Schulen nicht ausgestattet und die Aufsicht bei solchen Klassenarbeiten ist schwer zu realisieren. Für die Aufgabe zur Tabellenkalkulation wird somit hier nur eine innermathematische Aufgabenstellung angeführt:

Aufgabe 40

Die beiden folgenden Tabellen stellen einen Ausschnitt aus einer Tabellenkalkulation dar. In der linken Tabelle werden die Zellen **A3** und **B2** in alle darunter liegenden Zellen kopiert.

a) Trage in der linken Tabelle die beim Kopieren entstehenden Formeln ein.

b) Trage in die rechte Tabelle die entstehenden Zahlenwerte ein.

c) Zeichne ein Punktdiagramm aus den entstehenden Daten.

	A	B
1	“x	“f(x)=1/x
2	-1	=1/A2
3	=A2+0,1	
4		
5		
6		
7		

	A	B
1	“x	“f(x)=1/x
2	-1	
3		
4		
5		
6		
7		

Tabelle 66: Aufgabe zur Anwendung der Tabellenkalkulation für eine Lernerfolgskontrolle

5. Vom Funktionsgraphen zum Funktionsterm

5.1 Didaktische Überlegungen

Die Schülerinnen und Schüler sollen im Laufe der Beschäftigung mit Funktionen lernen, Funktionsterme in mathematischer Schreibweise mit Hilfe von Variablen zu beschreiben. Die Eingabe in die Tabellenkalkulation ist dabei eine Zwischenstufe, die in sich sinnvoll ist und zum gewünschten Ziel führt. Für die Schülerinnen und Schüler ist es daher nicht einsichtig, warum sie die Funktionsterme mit ‘x‘ aufstellen sollen und vereinfachen sollen. Aus Sicht der Mathematik ist die Kenntnis verschiedener Funktionstypen Selbstzweck, dies allein kann kaum überzeugen. Auch die Fähigkeit, bestimmte Funktionstypen einfach zeichnen zu können, wenn man gute Kenntnisse von Funktionen hat, bezieht seine Gründe nur aus der inneren Systematik des Mathematikunterrichts und macht kaum die Notwendigkeit der genannten Fähigkeit klar. Ein anderes Motiv kann den Schülerinnen und Schülern leichter nahe gebracht werden, nämlich die Vereinfachung beim Tippen, wenn die Tabellen mit der Hand erstellt werden, und das leichtere Nachvollziehen von Rechnungen, wenn diese kompakter dargestellt werden. Wichtiger ist es jedoch die Einsicht, dass es sinnvoll ist, ein gewisses Repertoire von Funktionen zu kennen, um durch Funktionen beschriebene Situationen klassifizieren zu können. Hierfür soll Kapitel 5.2 ein Motiv bieten.

5.2 Messreihen interpretieren und reproduzieren

Wenn man eine Reihe von Messwerten in der Physik betrachtet, wird man diese mit Hilfe der bekannten Funktionen einordnen, um eine Gesetzmäßigkeit zu beschreiben. Dazu muss der Funktionsterm in standardisierter und kompakter Form vorliegen. Dies wird anhand von entsprechenden Beispielen schnell klar. Die Beispiele dürfen dabei jedoch nicht trivial, also keine linearen Zusammenhänge sein.

Die folgenden oder vergleichbare Probleme sollten an geeigneter Stelle in den Unterricht eingeführt werden, wenn die benötigten Funktionen bekannt sind, Schülerinnen und Schüler die Systematik der Tabellen aufgefasst haben und der Übergang von der Formulierung mit Hilfe der Tabellenkalkulation zur mathematischen Sprache ansteht.

Aufgabe 41

In Kooperation mit der Physiklehrkraft wird eine Messreihe s(t) zum freien Fall aufgenommen. Sehr gut eignet sich hierfür ein Zeitmarkengeber, der mit der Netzfrequenz von 50 Hz Punkte auf einen speziellen Papierstreifen markiert (PHYWE 11607.00). Diese Streifen können sehr schnell hergestellt werden und jeder Schülerin und jedem Schüler zur Auswertung in die Hand gegeben werden. Bei sorgfältiger Versuchsdurchführung erhält man sehr gute Messergebnisse, bei denen jedoch die für Experimente typischen Streuungen vorhanden sind. Trägt man diese Messergebnisse in ein Koordinatensystem ein, sieht man die Parabel. Die Aufnahme der Messergebnisse direkt in die Tabellenkalkulation ermöglicht es zu versuchen, in einer weiteren Spalte eine Parabel durch die Punkte zu legen, indem man den Faktor vor dem x^2 variiert.

Aufgabe 42

Eine weitere nicht lineare Messung ist der zweiseitige Hebel, wenn das Drehmoment auf der einen Seite konstant bleibt und die für das Gleichgewicht benötigte Kraft in Abhängigkeit vom Hebelarm ermittelt wird.

Aufgabe 43

Zwischen der (mittleren) Bahngeschwindigkeit der Planeten unseres Sonnensystems und ihrem (mittleren) Abstand von der Sonne besteht der folgende Zusammenhang. Findet man eine Funktionsvorschrift, die den Zusammenhang x(v) beschreibt? (nach [4])

	Merkur	Venus	Erde	Mars	Jupiter	Saturn	Uranus	Neptun	Pluto
x in 1000 km	58344	107712	149600	227392	777920	1428680	2872320	4502960	5909200
v in km/s	47,6	35	29,8	24,1	13	9,6	6,8	5,4	4,7

Tabelle 67: Planetengeschwindigkeiten und Bahnradien

Hier ist x proportional zu $\frac{1}{v^2}$.

5.3 Phänomenologie von Funktionen

Bei der Untersuchung von Funktionsklassen (Geraden, Parabeln, Exponentialfunktionen) sollte die Verwendung der Tabellenkalkulation die Anfertigung von Zeichnungen keineswegs ersetzen, für die Produktion großer Mengen von Anschauungsmaterial ist sie jedoch eine gute Ergänzung, die zudem die Verwendung von Parametern motiviert.

	A	B	C	D	E	F
1		a*x+b	a*x^2+bx+c	a*b^x	Weite =	0,1
2	a=	-0,5	1	3		
3	b=	2	2	0,5		
4	c=		-3			
5	x	Gerade	Parabel	Exp-Fkt		
6	-4	=B$2*A6+B$3	=C$2*A6^2+C$3*A6+C$4	=D$2*D$3^A6		
7	=A6+F$1	=B$2*A7+B$3	=C$2*A7^2+C$3*A7+C$4	=D$2*D$3^A7		

Tabelle 68: Funktionen in Abhängigkeit von Parametern



Abb. 28: Funktionszeichnungen zu Tabelle 68

Bei dieser Untersuchung von Funktionsscharen ist eine genaue Dokumentation der Ergebnisse durch die Schülerinnen und Schüler besonders wichtig, da sonst nur viele Bilder angesehen werden und nichts im Gedächtnis bleibt. Möglich ist hier die systematische Untersuchung der Wirkung der einzelnen Parameter mit textlicher Formulierung der Ergebnisse oder die geometrische Vorgabe von bestimmten Funktionen, die dann durch Probieren reproduziert werden. Hierbei kann es sich als schwierig erweisen, dass die Maßstäbe der Koordinatenachsen durch die Tabellenkalkulation automatisch festgelegt werden, so dass die Schülerinnen und Schüler beim Vergleich von Vorgabe und Reproduktion auf die Einheiten der Achsen achten müssen. Diesem Problem kann durch eine feste Skalierung der Achsen (in Excel: Doppelklick auf die jeweilige Achse, Menükarte ‚Skalieren) begegnet werden.

6 Lösen von Gleichungen durch Probieren

6.1 Didaktische Überlegungen

Für das Lösen von Gleichungen dient die Tabellenkalkulation nur als didaktisches Mittel, um Schülerinnen und Schülern das Verständnis für die Aufgabenstellung und den Sinn und die Bedeutung von Gleichungen näher zu bringen. Das Lösen von Gleichungen geschieht in einer Tabellenkalkulation numerisch über den Vergleich zweier Spalten, die sich beide auf dieselbe 'x-Spalte' beziehen oder über numerische Verfahren wie das Newtonverfahren. Dieses Vorgehen unterstützt den Variablenbegriff und stellt eine wichtige Verbindung zum Funktionsbegriff her: eine Gleichung ist das Vergleichen zweier Funktionen. Dies wird später in die andere Richtung verwendet: das Gleichsetzen zweier Funktionen als Ansatz zum Finden von Schnittpunkten.

Mit Hilfe der Tabellenkalkulation wird also die Behandlung der Lösung von Gleichungen eingeführt. Es bietet sich an, dies erst nach den Untersuchungen von Funktionen durchzuführen, um auf die dabei auftretenden Beispiele zurückgreifen zu können. Man kommt dann zu der Frage, ob bestimmte Aufgaben nicht einfacher ohne Tabellenkalkulation mit Hilfe analytischer Verfahren zu lösen sind. Dabei sollten wieder nicht zu einfache Aufgaben der Lernanlass sein, und die Schülerinnen und Schüler sollten deutlich erfahren, dass die Betrachtung einfacherer innermathematischer Beispiele mit dem Ziel durchgeführt wird, auch das komplexere Phänomen zu lösen.

Für die Behandlung von Textaufgaben gilt das Gleiche, wie schon in Kap. 4.2 beschrieben. Vor der Formulierung sollten die Schülerinnen und Schüler die Aufgabe mit Zahlenbeispielen durchrechnen, um die zugrunde liegenden Rechenwege zu finden. Dieses Vorgehen findet sich wiederum beim Lösen einer Gleichung mit der Tabellenkalkulation als systematisches Vorgehen wieder, so dass die Tabellenkalkulation als denkstrukturbildendes Element seine Berechtigung hat. Haben die Schülerinnen und Schüler diesen Zugang zu Textaufgaben gefunden, sollten solche Aufgaben im Mittelpunkt der Übungsphasen stehen und innermathematische Übungsaufgaben weitgehend ersetzen, um die Fähigkeit zu festigen, aus Texten die mathematische Formulierung zu entwickeln. Dann sind auch in den folgenden Jahren problemorientierte Unterrichtsgänge möglich.

6.2 Beispielaufgaben

Einige der bereits dargestellten Aufgaben (Nr. 23, 30, 33, 34, 35, 36 und 39) lassen sich als Suche nach der Lösung von Gleichungen auffassen. Daher ist hier nur ein weiteres Anwendungsbeispiel angeführt. Prinzipiell eignen sich alle Textaufgaben für diese Herangehensweise.

Aufgabe 44

Eine Maschine kostet 20 000 DM. Der Wartungsvertrag kostet im ersten Jahr 900 DM, in den folgenden Jahren jedes Jahr 400 DM mehr aufgrund des steigenden Ersatzteilbedarfes. Die Anschaffung einer neuen Maschine wird jährlich 2% teurer. Ab wann lohnt es sich, eine neue Maschine zu kaufen, wenn die Zinsen für die Finanzierung 6% betragen und jährlich 15% getilgt werden (nach [4])?

	A	B	C	D	E	F
1	Inflation =	0,02	Zinsen =	0,06	Tilgung =	0,15
2	Jahr	Wartung	Neupreis	Kosten		
3	1	900	20000	=C3*(D$1+F$1)		
4	=A3+1	=B3+400	=C3*(1+B$1)	=C4*(D$1+F$1)		
5	=A4+1	=B4+400	=C4*(1+B$1)	=C5*(D$1+F$1)		
6	=A5+1	=B5+400	=C5*(1+B$1)	=C6*(D$1+F$1)		
7	=A6+1	=B6+400	=C6*(1+B$1)	=C7*(D$1+F$1)		

Tabelle 69: Kalkulation von Maschinenkosten

Zu vergleichen sind die Spalten B und D. Nach etwa 12 Jahren sind die Kosten für eine neue Maschine geringer als die Wartungskosten.

Aufgabe 45

Auch historische Aufgaben können mit moderner Technologie ihren angestaubten Charme wieder entfalten:
Dem Mönch Alkuin (735 bis 804), Berater Karls des Großen, wird eine Sammlung arithmetischer Aufgaben zugeschrieben, die stark an die Inder anklingt. Eine Aufgabe lautet:
Ein Wanderer trifft mit Schülern zusammen und fragt sie: Wie viel seid ihr in der Schule? Da antwortet einer von ihnen: Nimm unsere Zahl doppelt, multipliziere sie mit 3 und dividiere durch 4. Rechnest du noch mich dazu, sind es im ganzen 100 (aus [3]).

	A	B	C	D	E	F	G
1	Schülerzahl	mal 2	mal 3	durch 4	plus 1		
2	1	=A2*2	=B2*3	=C2/4	=D2+1		
3	=A2+1	=A3*2	=B3*3	=C3/4	=D3+1		
4	=A3+1	=A4*2	=B4*3	=C4/4	=D4+1		
5	=A4+1	=A5*2	=B5*3	=C5/4	=D5+1		
6	=A5+1	=A6*2	=B6*3	=C6/4	=D6+1		
7	=A6+1	=A7*2	=B7*3	=C7/4	=D7+1		

Tabelle 70: Historische Textaufgabe

Gerade diese einfachen Aufgaben zeigen, wie das Zusammenfassen von Rechenschritten die Arbeit erleichtern würde. Ein solches Vorgehen soll Schülerinnen und Schülern den Weg zu den analytischen Lösungen weisen.

7 Weitere Anwendungen im Unterricht

Ist Schülerinnen und Schülern der Umgang mit einer Tabellenkalkulation bekannt, gibt es zahlreiche Situationen im Unterricht, in denen die Anwendung des Gelernten durch die Verwendung dieser Software dadurch unterstützt werden kann, dass behandelte Algorithmen tatsächlich programmiert werden müssen. Gerade bei Verfahren der numerischen Mathematik ist dabei die Verwendung der Tabellenkalkulation sehr angenehm, da man sich auf die numerischen Schritte konzentrieren kann und sich nicht mit Programmstrukturen und Eingabe/Ausgaberoutinen befassen muss.

Aufgabe 46

Es ist $\sqrt{2}$ mit Hilfe des Heronverfahrens zu bestimmen:

	A	B	C	D	E	F	G
1	Wurzel von	2					
2	1. Seite	2. Seite					
3	1	=B$1/A3					
4	=(A3+B3)/2	=B$1/A4					
5	=(A4+B4)/2	=B$1/A5					
6	=(A5+B5)/2	=B$1/A6					
7	=(A6+B6)/2	=B$1/A7					

Tabelle 71: Heronverfahren

Aufgabe 47

Eine Nullstelle der Funktion $f(x) = x - \cos(x)$ soll mit dem Newtonverfahren ermittelt werden.

	A	B	C	D	E	F	G
1	Startwert	3					
2	x_n	Funktion	Ableitung	x_{n+1}			
3	=B1	=COS(A3)-A3	=-SIN(A3)-1	=A3-B3/C3			
4	=D3	=COS(A4)-A4	=-SIN(A4)-1	=A4-B4/C4			
5	=D4	=COS(A5)-A5	=-SIN(A5)-1	=A5-B5/C5			
6	=D5	=COS(A6)-A6	=-SIN(A6)-1	=A6-B6/C6			
7	=D6	=COS(A7)-A7	=-SIN(A7)-1	=A7-B7/C7			

Tabelle 72: Newtonverfahren

Die Ableitung muss hier natürlich per Hand für jede Funktion neu bestimmt werden.

Aufgabe 48

Für die Wahrscheinlichkeitsrechnung wird ein PASCAL-Dreieck benötigt:

	A	B	C	D	E	F	G
1	1						
2	1	1					
3	1	=A2+B2	1				
4	1	=A3+B3	=B3+C3	1			
5	1	=A4+B4	=B4+C4	=C4+D4	1		
6	1	=A5+B5	=B5+C5	=C5+D5	=D5+E5	1	
7	1	=A6+B6	=B6+C6	=C6+D6	=D6+E6	=E6+F6	1

Tabelle 73: PASCAL-Dreieck

Mit diesem Verfahren können leicht sehr große PASCAL-Dreiecke hergestellt werden, die mit Hilfe der Tabellenkalkulation auch einfach formatiert und ausgegeben werden könne.

Aufgabe 49

Bei einer Weltumsegelung soll der tägliche Mindestbedarf an den Vitaminen A, C und K durch Vitaminpräparate gedeckt werden. In Frage kommen Präparate P_1 und P_2 mit den folgenden Merkmalen (aus [9]):

	Vitamin A	Vitamin C	Vitamin K	Preis
Tagesbedarf	1,5 mg	150 mg	20 mg	
Inhalt in P_1	0,1 mg	20 mg	1 mg	0,10 DM
Inhalt in P_2	0,15 mg	10 mg	4 mg	0,20 DM

Tabelle 74: Aufgabendaten zum Vitaminbedarf

	A	B	C	D	E	F	G	H	I
1		Vitamin A	Vitamin C	Vitamin K	Preis				
2	Ziel-größen	1,5	150	20					
3	P_1	0,1	20	1	0,1				
4	P_2	0,15	10	4	0,2				
5					Bedarf	=A4			
6	=A3	=B1	=C1	=D1	=B1	=C1	=D1	Anzahl	Kosten
7	0	=$A7*B$3	=$A7*C$3	=$A7*D$3	=(B$2-B7)/B$4	=(C$2-C7)/C$4	=(D$2-D7)/D$4	=MAX(E7;F7;G7)	=A7*E$3+H7*E$4

Tabelle 75: Optimierung des Vitaminangebotes

Diese Tabelle kann für alle vergleichbaren Optimierungsaufgaben verwendet werden, wobei nur in die Felder A1 bis E4 die jeweils zur Aufgabe gehörenden Daten und Überschriften erhalten müssen. Damit die Überschriften in den Zeilen 5 und 6 dann auch stimmen, wurden sie durch Verweise auf die anderen Überschriften realisiert. So muss man für die folgende Aufgabe 50 nur die Einträge aus Tabelle 76 in die Felder A1 bis E4 der Tabellenkalkulation übernehmen:

Aufgabe 50

Die Gärtnerei Timm lässt ihren Boden auf notwendige Düngemittel untersuchen. Für die Düngung stehen zwei Volldünger mit folgenden Merkmalen zur Verfügung (aus [9]):

	Kali	Phosphor	Stickstoff	Preis
Düngemittelbedarf	1200 kg	1200 kg	1500 kg	
Dünger 1	400 kg	300 kg	300 kg	120,00 DM
Dünger 2	200 kg	300 kg	600 kg	80,00 DM

Tabelle 76: Optimierter Düngemitteleinsatz

Zur Lösung kann Tabelle 76 verwendet werden, es müssen nur die Daten analog übertragen werden.

Aufgabe 51

Das Lösen eines linearen Gleichungssystems mit drei Gleichungen und drei Unbekannten ist rechenintensiv und fehleranfällig. Daher soll der Algorithmus mit Hilfe einer Tabellenkalkulation durchgeführt werden.
Der Kern dieser Aufgabe besteht nicht darin, ein Gleichungssystem zu lösen, sondern das Rechenverfahren zu strukturieren und zu optimieren. Die hier vorgestellte Lösung kann erst am Ende einiger Versuche stehen.

	A	B	C	D	E	F	G	H	I	J
1	Lösen eines linearen Gleichungssystems mit 3 Unbekannten.									
2										
3	Nr. 1	1	x+	3	y+	4	z=	3	mal	=-B4*B5
4	Nr. 2	7	x+	4	y+	6	z=	-8	mal	=B3*B5
5	Nr. 3	3	x+	5	y+	7	z=	2	mal	=B4*B3
6										
7	Nr. 1a	=B3*$J3	x+	=D3*$J3	y+	=F3*$J3	z=	=H3*$J3		
8	Nr. 2a	=B4*$J4	x+	=D4*$J4	y+	=F4*$J4	z=	=H4*$J4		
9	Nr. 3a	=B5*$J5	x+	=D5*$J5	y+	=F5*$J5	z=	=H5*$J5		
10										
11	Nr. 1b=1a+2a	=B7+B8	x+	=D7+D8	y+	=F7+F8	z=	=H7+H8	mal	=-D12
12	Nr. 2b=1a+3a	=B7+B9	x+	=D7+D9	y+	=F7+F9	z=	=H7+H9	mal	=D11
13										
14	Nr. 1c			=D11*$J11	y+	=F11*$J11	z=	=H11*$J11		
15	Nr. 2c			=D12*$J12	y+	=F12*$J12	z=	=H12*$J12		
16										
17	Nr. 1d=1c+2c			=D14+D15	y+	=F14+F15	z=	=H14+H15	durch	=F17
18										
19							z=	=H17/F17		
20										
21	Nr. 1c			=D14	y+	=F14*H19	=	=H14	minus	=F21
22										
23				=D21	y		=	=H21-J21	durch	=D23
24										

	A	B	C	D	E	F	G	H	I	J
25					y		=	=H23/J23		
26										
27	Nr. 1	=B3	x+	=D3*H25	+	=F3*H19	=	=H3	minus	=D27+F27
28										
29		=B27	x				=	=H27-J27	durch	=B29
30										
31			x				=	=H29/J29		

Tabelle 77: Lösen eines linearen Gleichungssystems

In Spalte J stehen jeweils die Zahlen, mit denen auf den Gleichungen operiert wird. Diese sind zum Ende der Aufgabe automatisiert, sollten jedoch zunächst als konkrete Zahlen von den Schülern eingegeben werden, damit sie das Rechenverfahren schrittweise besser verstehen. Bei einfachen Aufgaben werden die Faktoren deutlich kleiner ausfallen, als die per Formel berechneten. Insbesondere die Faktoren in J3 bis J5 werden erst so gewählt wie hier berechnet, wenn die Zahlen in dem Gleichungssystem in hohem Maße teilerfremd sind. Hier helfen also letztlich schwere Aufgaben, den Algorithmus vollständig zu implementieren.

In der vorliegenden Form versagt die Tabelle, wenn an ungünstigen Stellen eine Null auftritt (z.B. in B4). Dieser Mangel kann durch geeignete Abfragen mit dem Wenn-Befehl behoben werden, wozu leistungsstarke Schülerinnen und Schüler durchaus in der Lage sein sollten. Dieses Problem ist ein Anlass, die Rolle von Spezialfällen und deren systematische Untersuchung zu diskutieren.

Aufgabe 52

In der Oberstufe werden lineare Gleichungssysteme mit Matrizen gelöst. Dies geschieht in Excel mit den Befehlen MINV (inverse Matrix bilden) und MMULT (Matrixmultiplikation).

	A	B	C	D	E	F	G	H	I	J	K	L
1	Lineares Gleichungssystem mit Matrizenfunktionen lösen											
2									Als Matrix schreiben			
3	1	x+	3	y+	4	z=	3		=A3	=C3	=E3	=G3
4	7	x+	4	y+	6	z=	-8		=A4	=C4	=E4	=G4
5	3	x+	5	y+	7	z=	2		=A5	=C5	=E5	=G5
6												
7									Inverse der 3*3 Matrix bilden			
8									=MINV(I3:K5)	=MINV(I3:K5)	=MINV(I3:K5)	=L3
9									=MINV(I3:K5)	=MINV(I3:K5)	=MINV(I3:K5)	=L4
10									=MINV(I3:K5)	=MINV(I3:K5)	=MINV(I3:K5)	=L5
11												
12									Multiplizieren			
13									x=	=MMULT(I8:K10;L8:L10)		
14									y=	=MMULT(I8:K10;L8:L10)		
15									z=	=MMULT(I8:K10;L8:L10)		

Tabelle 78: Einsatz von Matrixoperationen

Die Eingabe in den Feldern I8 bis K10 erfolgt in einem Schritt: Der Bereich wird markiert, die Formel wird eingegeben und der Befehl mit `STRG-UMSCHALT-RETURN` abgeschlossen. Diese gilt auch für den Bereich J13 bis J15. Excel schreibt dann zusätzlich geschweifte Klammern um den Befehl, um ihn als Matrixbefehl zu kennzeichnen. Leider ist der Befehlssatz für Matrizen nicht vollständig, es fehlt beispielsweise die Addition von Vektoren.

Aufgabe 53

Das Zeichnen von Elementen aus der fraktalen Geometrie kann mit der Tabellenkalkulation erfolgen.

	A	B	C
1	0,5	0	0
2	0	0,5	0,5
3			
4	-0,5	0	1
5	0	0,5	0
6			
7	0	-0,5	0,5
8	0,5	0	0
9			
10	0	0	=ZUFALLSZAHL()
11	=WENN(C10<0,33; ($A10*$A$1+$B10*$B$1+$C$1); WENN(C10<0,66; $A10*$A$4+$B10*$B$4+$C$4; $A10*$A$7+$B10*$B$7+$C$7))	=WENN(C10<0,33; ($A10*$A$2+$B10*$B$2+$C$2); WENN(C10<0,66; $A10*$A$5+$B10*$B$5+$C$5; $A10*$A$8+$B10*$B$8+$C$8))	=ZUFALLSZAHL()
12	=WENN(C11<0,33; ($A11*$A$1+$B11*$B$1+$C$1); WENN(C11<0,66; $A11*$A$4+$B11*$B$4+$C$4; $A11*$A$7+$B11*$B$7+$C$7))	=WENN(C11<0,33; ($A11*$A$2+$B11*$B$2+$C$2); WENN(C11<0,66; $A11*$A$5+$B11*$B$5+$C$5; $A11*$A$8+$B11*$B$8+$C$8))	=ZUFALLSZAHL()

Tabelle 79: Iteriertes Funktionensystem

Gezeichnet werden die Spalten A und B ab Zeile 11 als Punktdiagramm, wobei Spalte A als x-Achsenbeschriftung dient. In den Feldern A1 bis C8 stehen die drei Abbildungen. Es können auf einfache Weise mehr Abbildungen verwendet werden. Die Zufallszahl entscheidet, welche der drei Abbildungen verwendet wird. Man darf sie nicht in die Formeln mit aufnehmen, da sie sonst bei jeder Abfrage, das heißt in Zeile 11, jedesmal wenn statt C10 ‚ZUFALLSZAHL()' aufträte, neu erzeugt würde. Sie muss aber innerhalb einer Zeile konstant sein. Es empfiehlt sich, in der Tabellenkalkulation die Option zum automatischen Neuberechnen auszuschalten, während man neue Zahlen für die Abbildung eingibt, da sonst nach jeder Zahl neu gerechnet wird, was bei langsamen Rechnern sehr störend ist. Das Auslösen einer Neuberechnung ohne Änderung der Abbildungen führt dazu, dass die Zufallszahlen neu erzeugt werden, ohne dass sich die Geometrie des Fraktals ändert! Zum mathematischen Hintergrund und zur Einbettung in den Unterricht dient [5].



Abb. 29: Fraktal aus der Familie der Sierpinski-Verwandten

Aufgabe 54

Ebenso wie Fraktale kann man auch einen Feigenbaum mit Hilfe einer Tabellenkalkulation erzeugen:

	A	B	C	D	...
1	-2	=A1+0,05	=B1+0,05	=C1+0,05	
2	0,1	0,1	0,1	0,1	
3	=A2*A2+A$1	=B2*B2+B$1	=C2*C2+C$1	=D2*D2+D$1	
4	=A3*A3+A$1	=B3*B3+B$1	=C3*C3+C$1	=D3*D3+D$1	
5	=A4*A4+A$1	=B4*B4+B$1	=C4*C4+C$1	=D4*D4+D$1	
100	=A1	=B1	=C1	=D1	
101	=A99*A99+A$1	=B99*B99+B$1	=C99*C99+C$1	=D99*D99+D$1	
102	=A101*A101+A$1	=B101*B101+B$1	=C101*C101+C$1	=D101*D101+D$1	
103	=A102*A102+A$1	=B102*B102+B$1	=C102*C102+C$1	=D102*D102+D$1	
104	=A103*A103+A$1	=B103*B103+B$1	=C103*C103+C$1	=D103*D103+D$1	
105					

Tabelle 80: Tabelle zur Erzeugung eines Feigenbaumdiagramms zu $f_c(x) = x^2 + c$

Gezeichnet werden die Zellen A100 bis AT205 als Punktdiagramm ohne Verbindungslinie, wobei Daten in Zeilen dargestellt werden. Die Zeile 100 dient zur Skalierung der x-Achse, in ihr steht jeweils der Wert von $c \in [-2;0,25]$. Die Herstellung des Feigenbaumdiagramms geht auf diese Weise sehr schnell. Langwierig war nur die Überarbeitung der fertigen Grafik, da Excel einen automatischen Wechsel der Punktdarstellung in jeder Zeile vornimmt. Dies liefert auf dem Bildschirm interessante Effekte und ist nur für den Druck ungeeignet.



Abb. 30: Feigenbaumdiagramm

Aufgabe 55

Die Spirale aus Abb. 31 soll gezeichnet werden (nach [7]).

	A	B	C	D
1	Startlänge	5		
2	Verkürzungs-faktor	0,9	Punkt-x	Punkt-y
3		Start:	0	5
4	Länge	Richtung		
5	=B1	0	=C3	=D3
6	=A5*B$2	=REST(B5+1;4)	=WENN(B6=1;C5+A6; WENN(B6=3;C5-A6;C5))	=WENN(B6=2;D5-A6; WENN(B6=0;D5+A6;D5))
7	=A6*B$2	=REST(B6+1;4)	=WENN(B7=1;C6+A7; WENN(B7=3;C6-A7;C6))	=WENN(B7=2;D6-A7; WENN(B7=0;D6+A7;D6))

Tabelle 81: Eckige Spirale

Die Länge der Spirale lässt sich bei Bedarf ebenso berechnen, um Zahlenmaterial für die geometrische Reihe zu erhalten. In Zeile B wird eine zyklische Zahlenfolge 0-1-2-3 erzeugt, die die wechselnde Richtung der Kurve steuert.



Abb. 31: Spirale

Aufgabe 56

Rollkurven können eine interessantes Problemfeld im Unterricht der zehnten Klasse oder im Ergänzungskurs sein, insbesondere wenn man sich auf die mittels Spirographen erzeugte Rollkurven konzentriert [6].

Diese Kurven können nach der mathematischen Analyse auch mit einer Tabellenkalkulation hergestellt werden.

	A	B	C	D	E	F	G	H
1	r1=	105			Schrittweite =	0,1		
2	r2=	63						
3	p=	0,8						
4	r1-r2=	=B1-B2						
5	r2*p=	=B2*B3						
6								
7	alpha	beta	mx	my	ax	ay	zx	zy
8	0	=-B$1*A8/B$2	=B$4*COS(A8)	=B$4*SIN(A8)	=B$5*COS(A8+B8)	=B$5*SIN(A8+B8)	=C8+E8	=D8+F8
9	=A8+F$1	=-B$1*A9/B$2	=B$4*COS(A9)	=B$4*SIN(A9)	=B$5*COS(A9+B9)	=B$5*SIN(A9+B9)	=C9+E9	=D9+F9

Tabelle 82: Rollkurven

Die Radien in B1 und B2 sind die der beteiligten Kreise, wobei der kleinere im größeren rollt. B3 besagt, wie weit der Zeichenstift , als Faktor vor dem Radius vom Rand des inneren Kreises entfernt ist. In den Spalten C und D stehen die Koordinaten des Mittelpunktes des rollenden Rades und in den Spalten E und F die Koordinaten des Zeichenpunktes vom Mittelpunkt aus betrachtet. Gezeichnet werden die Spalten G und H.



Abb. 32: Spirographenkurve zu Tabelle 82.

Der Einsatz einer Tabellenkalkulation bietet sich darüber hinaus noch an, wenn beispielsweise im Gemeinschaftskundeunterricht Befragungen durchgeführt werden, die statistisch ausgewertet werden sollen. Ebenso können sämtliche Messungen im Physikunterricht mit Hilfe der Tabellenkalkulation schnell und genau ausgewertet und mit Grafiken versehen werden. Dabei können statistische Methoden eingesetzt werden, die ohne den Einsatz dieses Werkzeuges mit erheblichem Aufwand verbunden wären.

Aufgabe 57

Quadratische Regression zu den Messwerten aus Aufgabe 41. Auswertungen komplexerer Art sind ebenfalls einfach ausführbar.

	A	B	C	D	E	F
1	t	s	Wurzel(s)			
2	0	=9,81*A2^2/2	=WURZEL(B2)	=C2/A2		
3	=A2+0,02	0,002	=WURZEL(B3)	=C3/A3	=MITTELWERT(D3:D31)	=2*E3^2
4	=A3+0,02	0,007	=WURZEL(B4)	=C4/A4		
5	=A4+0,02	0,02	=WURZEL(B5)	=C5/A5		
6	=A5+0,02	0,035	=WURZEL(B6)	=C6/A6		
7	=A6+0,02	0,052	=WURZEL(B7)	=C7/A7		

Tabelle 83: Auswertung freier Fall

In F3 erscheint der aus dem Experiment erwachsende Schätzwert für die Erdbeschleunigung.

8 Beispiele für den Einsatz von Tabellenkalkulationen außerhalb der Schule

Durch die Möglichkeit, die Tabellenkalkulation in der Schule fächerübergreifend einzusetzen, wird ihre Verwendung zusätzlich zu den didaktischen Vorteilen sinnvoll. Die folgenden Beispiele sollen als Hintergrundinformation dienen, um Beispiele für den Einsatz von Tabellenkalkulation außerhalb der Schule zu haben.

- Beim Bau meines Hauses hat der Architekt sämtliche Berechnungen für den Materialverbrauch, die Ausschreibungsprüfung und die Überwachung der Zahlungsverpflichtungen und Rechnungen der Handwerker mit einer Tabellenkalkulation durchgeführt.
- Im privaten Bereich kann mit der Tabellenkalkulation der Überblick über regelmäßige Ausgaben und Einnahmen gut organisiert werden, wobei die Summen über Formeln berechnet werden (aus Platzgründen wurde hier auf die Zeilen- und Spaltenbezeichnungen verzichtet):

	Hypothek	HEW	Gas	Wasser	Telefon	Versicherungen	Beiträge	Summe	Gehalt	Saldo
Januar		86	95		80	739	20	1020	6274	5254
Februar		86	95	132	80	739	20	1152	6274	5122
März	7300	86	95		80	739	65	8365	6274	-2091
April		86	95	132	80	739	20	1152	6274	5122
Mai		86	95		80	739	20	1020	6274	5254
Juni	7300	91	95	132	80	739	65	8502	6274	-2228
Juli		91	95		80	739	20	1025	6274	5249
August		91	95	132	80	739	20	1157	6274	5117
September	7300	91	95		80	739	65	8370	6274	-2096
Oktober		91	95	242	80	739	20	1267	6274	5007
November		91	95		80	739	20	1025	6274	5249
Dezember	7300	91	95	132	80	739	65	8502	10000	1498
Summe	29200	1067	1140	902	960	8868	420	42557	79014	36457

Tabelle 84: Haushaltsübersicht

- In der Fachhochschule Hamburg (und vermutlich nicht nur dort) wird die Mittelverteilung auf die Fachbereiche nach einem Kennziffernsystem – abhängig von verschiedenen Parametern – mit einer Tabellenkalkulation durchgeführt. Das Kennziffernsystem selbst wurde auf der Grundlage umfangreicher Berechnungen aus bisherigen Haushaltsansätzen, Studentenzahlen, Professorenzahlen und anderen entwikkelt. Dies gelang erst fehlerfrei, als die Berechnungen mit einer Tabellenkalkulation durchgeführt wurden.
- Für Hypothekentilgungen sind umfangreiche Berechnungen erforderlich. Die Berechnung mit einer Tabellenkalkulation ist hier die einfachste Methode:

	A	B	C	D	E	F	G	H	I	J
1	Starttermin		=DATUM(1998;3;1)		Laufzeit	=A138	Jahre bis	=B138		
2	Hypothek		170000		Belastung	=H7	DM vierteljährlich			
3	Zinssatz		0,0613			=F2/3	DM monatlich			
4	Tilgung		0,07		Gesamtzahlung	=C138+I138				
5					Gesamtzins	=F4-C2		=WENN(C138<0;1;0)	Fehler wenn 0	
6		Jahr	Restschuld	Zinssatz	Zins	Tilg%	Tilg	Zahlung	Sum Zahl	Sum Tilg
7	0	=C1	=C2	=C3	=WENN(C7>0;C7*D7/4;0)	=C4	=H7-E7	=WENN(C7>0;$C$7*(D7+F7)/4;0)	=H7	=G7
8			=C7-G7	=D7	=WENN(C8>0;C8*D8/4;0)	=F7	=H8-E8	=WENN(C8>0;$C$7*(D8+F8)/4;0)	=WENN(C8>0;I7+H8;I7)	=J7+G8
9			=C8-G8	=D8	=WENN(C9>0;C9*D9/4;0)	=F8	=H9-E9	=WENN(C9>0;$C$7*(D9+F9)/4;0)	=WENN(C9>0;I8+H9;I8)	=J8+G9

Tabelle 85: Hypothekentilgung

In dieser Tabelle werden die Formeln bis Zeile 138 gefüllt. In den Spalten A und B wird die Zeit und der Monat in Vierteljahresschritten hochgezählt, was leider nicht so einfach funktioniert, wie es sein sollte. Die Formel funktioniert mit mehreren Abfragen, um Jahreswechsel korrekt zu erfassen und ist deshalb hier nicht abgedruckt. Dabei fragt diese Berechnung noch die Restschuld in Spalte C ab, um zu prüfen ob noch getilgt werden muss. Wenn nicht, bleibt das Datum stehen. Analog wird in den Spalten E, H und I abgefragt, so dass nach der vollständigen Tilgung die folgenden Zeilen sich nicht mehr ändern. Dadurch kann in F4 die Gesamtzahlung abgefragt werden. In H5 wird geprüft, ob die Tilgung in Zeile 138 schon ganz abgeschlossen ist, wenn nicht, wird eine 0 ausgegeben. Die notwendigen Eingaben geschehen in den Zellen C1 bis C4, in den anderen Zellen des Kopfes werden die resultierenden Kenndaten der Tilgung angeführt.

9 Anmerkungen zum Einsatz eines Computeralgebrasystems

Viele Schulen haben im Rahmen der Landeslizenz des Amtes für Schule (S 13/2) das Computeralgebrasystem DERIVE erworben. Wegen der zunehmenden Bedeutung solcher Werkzeuge im Zuge der Weiterentwicklung des Mathematikunterrichts halte ich dies für sehr begrüßenswert. Bei der Einführung in den Umgang mit Funktionen, Termen und Gleichungen sehe ich allerdings Vorteile einer Tabellenkalkulation gegenüber der Verwendung eines Computeralgebrasystems wie DERIVE.

1. DERIVE ist ein Mathematikerwerkzeug und hat daher nicht ganz die motivierende Wirkung wie ein im Arbeitsleben eingesetztes Werkzeug.
2. Selbst DERIVE als das am einfachsten zu bedienende Computeralgebrasystem ist in der Bedienung nicht so einfach wie moderne Tabellenkalkulationen. Als Einstiegssoftware ist die Tabellenkalkulation daher besser geeignet.
3. DERIVE setzt beim Anwender die Kenntnis von Variablen und Termen voraus, kann also nur schwer zur Einführung dieser Begriffe dienen. Insbesondere fehlt die in der Anfangsphase didaktisch positive ständige Konkretion des Variableninhaltes wie in den Zellen der Tabellenkalkulation.

DERIVE beherrscht Termumformungen und das Lösen von Gleichungen souverän. Diese Aktivitäten sollen die Schülerinnen und Schüler in den angesprochenen Klassenstufen jedoch erst lernen. Sowenig wie man Kindern die Grundrechenarten beibringt, indem man mit ihnen den Taschenrechner anwendet und sowenig wie man während des Lernens des ‚Malens' von Buchstaben eine Textverarbeitung einsetzt, sowenig sollte man Schülerinnen und Schülern ein Algebrasystem zum Erlernen der Algebra in die Hand geben. Andererseits ist der Einsatz von DERIVE nach dem gesicherten Erwerb der Fertigkeiten im Umgang mit Termen sehr sinnvoll, ebenso wie der Taschenrechner nach dem Erlernen der Grundrechenarten auf ⊗. Für den Einsatz im Unterricht müssen hierfür noch Materialien entwickelt werden. Wenn ein Computeralgebrasystem im Unterricht eingesetzt wird, werden sich etliche Lernziele im Unterricht verändern. Modellierungen von Sachverhalten werden noch wichtiger werden, mathematisch handwerkliche Tätigkeiten treten in den Hintergrund, ohne dass die entsprechenden Fähigkeiten ganz verschüttet werden dürfen. Denn: Ohne eine sichere Fertigkeit im Umgang mit Termen können auch keine Sachverhalte mit Termen formuliert werden. Die Verwendung der Tabellenkalkulation könnte eine Antwort auf dieses didaktische Problem sein.

10 Literatur

[1] Cukrowicz, J., Hahn/Dzewas, Analysis 1, Westermann 1992

[2] Kießwetter, K., Lesetexte Mathematik, unveröffentlicht

[3] Reidt, Wolf, Athen, Elemente der Mathematik, Band 1, Schroedel 1964

[4] Schmid, A., Schweizer, W., Lambacher Schweizer, Analysis Leistungskurs, Klett Verlag 1990

[5] Stender, P., Iteration, Handreichung der Schulbehörde Hamburg

[6] Stender, P., Mathematikunterricht mit dem Spirographen

[7] Tischel, G., Tobel, K., Analysis Grundkurs, Diesterweg Verlag 1991

[8] Tischel, G., Tobel, K., Spektrum der Mathematik 10, Diesterweg Verlag 1988

[9] Tischel, G., Tobel, K., Spektrum der Mathematik 8, Diesterweg Verlag 1986

	A	B	C	D	E	F
1						
2						
3						
4						
5						
6						
7						
8						
9						
10						